

Quick Programming Guide

(Understanding Custom Control)

Assembled by a collaboration between

Tech Support
&
Application Engineering

Table of Contents

Chapter 1: Introduction To Programming	4
Introduction To Programming	4
Chapter 2: Using the Custom Control Editor	6
Adding Inputs - Overview	7
Adding Outputs.....	9
Error Checking.....	11
Input Logic Operators.....	12
OR Status Detect 1 or more points	13
OR 1 or more points.....	14
OR NOT Status 1 or more points.....	15
AND 1 or more points.....	16
AND NOT 1 or more points	16
CONTINUE current input state	18
COMPLEMENT current input state	19
SAVE state to memory register	20
RECALL memory register.....	20
DELAY output actions	22
CYCLE OUTPUT ACTIONS	24
COUNT inputs.....	26
ANY n points in a point list.....	28
Compare analog.....	29
Compare DOW value.....	30
Output Logic Operators	31
Output Priority	32
HOLD 1 or more points	33
TRACK 1 or more points.....	35
SET 1 or more points	37
PULSE analog.....	39
DISCONNECT 1 or more MONITOR points	41
DISABLE 1 or more points	42
LED functions (FAST/SLOW/ON/OFF).....	44
RESET monitor or analog points.....	45
SET analog value	46
PRINT a message.....	47
CHLCODE Mapnet channel.....	48
DEVICE CODE smart sensors	49
SET ALARM THRESHOLD smart sensors.....	50
SET RATE OF RISE heat sensors.....	52
DISARM 1 or more points.....	53
Signal Coding Overview.....	54
MARCH Code control points	58
PNIS Code control points	60
CALIFORNIA Code control points.....	62
CODE (generic) control points	64
UPLAY or UPLAYQ audio messages.....	65
UROUTE audio	68

USILENCE audio channel.....	69
USPEAKER circuits on/off.....	70
UPNISV voice code speaker circuits.....	71
UPNISM message code speaker circuits.....	73
QCODE.....	75
UVOLUME.....	76
UPNIST tone code speaker circuits.....	77
MARCH60.....	78
MARCH120.....	79
TEMPORAL.....	80
MARCH20.....	81
HPULSE.....	82
TCODE on a list of TrueAlert points.....	83
ACKNOWLEDGE TBL/SUP/FIRE/PRI2.....	84
 Chapter 3: Additional Coding.....	 85
Audio Coding.....	85
DC Coding.....	86
 Chapter 4: Example Programming.....	 87
AHU Staggered Start.....	87
COUNT Example Program.....	90
 Index.....	 91

Chapter 1: Introduction To Programming

Introduction To Programming

The operation of a 4100U panel is controlled mainly by the use of SMPL. SMPL is a programming language designed specifically for monitoring input conditions and activating outputs.

The number of programs per a job differs according to which revision is being used. In rev 10, 11 and some of the rev 12 programmers, there can be up to eight (8) programs per job. In programmers rev 12.04.01 and beyond, there can be up to fifty-six (56) programs per job.

The first three programs are reserved for the system and are used for standard system operations. Programs with one or more equations are initially on and can be turned on and off by system pseudos.

Program Tip: Often programs are separated by use or function. If all the AHU equations are in one program, we can turn OFF that function by turning OFF the pseudo that controls that program.

<u>Program</u>	<u>System Pseudo</u>
System Default	P16 (Not recommended to turn this pseudo off)
Audio Default	P17 (Not recommended to turn this pseudo off)
System Options (coding)	P18 (Not recommended to turn this pseudo off)
User Program 3	P19
User Program 4	P20
User Program 5	P21
User Program 6	P22
User Program 7	P23

User Programs 8 through 55 will default to P19. The System Pseudo points for these programs can be user assigned, starting at user pseudo P512. If any system pseudo is left as the default P19, then when P19 is turn off, it will turn off Program 3 and any other program from 8-55 that is using that same pseudo.

Programs are made up of several equations. The basic structure of a SMPL equation is as follows:

Equation No: **Label:** <40 character label>

COMMENTS:

<User input describing what this equation was intended to do>

[END-OF COMMENTS]

INPUTS:

<Select Opcode Type > <Select Input **Qualifier**> <Select Item>

<Select Point>

[END-OF INPUTS]

OUTPUTS:

<Select Opcode Type > <Select Item> <Set/Reset Priority>

<Select Point>

[END-OF OUTPUTS]

END:

Where:

Input operation is a **selection** made from the **Inputs List**
Select Opcode Type is a selection made from the List.
Select Input Qualifier is a AND/OR/NOT selection
Select Item Is a select Input Qualifier from a List
Select Point Is select a point from a Tag List

Output operation is a selection made from the **Outputs List**.
Select Opcode Type Output selection from a List
Select Item ON/OFF/PRI
Set/Reset Priority Set Priority=X Reset Priority=X ***SEE NOTE 1**
Select Point Select a point from a Tag List.


NOTE 1: The Set/Reset Priority are dependent of the status of the input and the number value of the priority. As long as the input is true the output will be sitting at the set priority. When the input goes false the output now toggles to the Reset Priority.(THE OUTPUT DOES NOT CHANGE ,THE PRIORITY DOES) The number value of the priority works as follows. The lower the number the higher the priority.

Chapter 2: Using the Custom Control Editor

Custom Control is an option selected from Custom Control Tab of the 4100U Programming Unit. The CCE is a menu driven editor designed to generate only valid SMPL equations.

The CCE screen is divided into three sections, the custom control programs, equations and equation comment.

Each custom control program can be expanded to show all equations with in that group. Selecting a specific equation will display the contents in the equation and equation contents windows.



Key points to remember:

- Equations can be manipulated by:
 - Clicking into equation screen and using the available radio buttons to **Add**, **Edit** or access the **Taglist**.
 - By typing over its contents, or by exporting and using an external editor.
 - Windows cut, copy and paste functions also work.
- To add an equation, right click in the program tree and select **Add Equation**.
- To delete an equation, right click on the equation in the program tree and selecting **Delete Equation**.
- A good practice to follow is to label all of your equations, to serve as a reminder of the operations they contain.
- Equations are numbered sequentially, beginning with one.
- The maximum number of equations per program is limited by the memory available. The more input/output operations in an equation, the more memory the equation will occupy.
- All equations must have an INPUT and an OUTPUT. As an example, if an equation doesn't have an input, the output is then a continuation of the previous equation.

To create new equations:

- 1.) Click the Custom Control tab in the programmer.
- 2.) Right click on the program (3-7 or 3-55) you wish to add the equation to.
- 3.) Click on Add Equation.
- 4.) Type in a label that makes sense.
- 5.) Click O.K.

The first part of the equation is the comment. A comment line has no specified character limit but should be kept at a minimum to conserve memory. For most equations the label is an adequate description of the equation's function and the comment can be left blank.

Adding Inputs - Overview

Key points to remember:


- Typically, there are four major parts to an input, though this number can vary depending on the Select Opcode Type field entry.
- If the input goes true the output operation is executed.

To add an Input statement:

- Click on [END INPUTS]
- Click the **ADD** button
- Use the Arrow button to the left of the **INPUT** field to choose your Opcode Type
- Click the **NEXT** button




- Click on the appropriate Qualifier(s)
- Click the **NEXT** button



- Use the Arrow button to the left of the **Opcode** field to choose your Opcode
- Click the **NEXT** button



- Type in the Point you wish to use. If you need to, use the **TAGLIST** button to find the point(s)
- Click the **FINISH** button



This would be a typical smoke detector input:

```
[INPUTS]  
  STATUS DETECT  
  M1-1-0 | PHOTO | SMOKE | TYPICAL SMOKE DETECTOR  
[END INPUTS]  
[OUTPUTS]  
[END OUTPUTS]
```

Adding Outputs

Key points to remember:

- The output side of an equation is an instruction, selected from a Custom Control menu, indicating the action to be taken when the input goes true. A common output action is the activation of relays or signals
- Most outputs have **priorities**, read the sections on “Output Logic Operators” & “Output Priority”.

To add an Output statement:


- Click on [END OUTPUTS]
 - Click the **ADD** button
 - Use the Arrow button to the left of the **OUTPUT** field to choose your Opcode Type
 - Click the **NEXT** button
-
- Use the Arrow button to the field to choose your Qualifier
 - Click the **NEXT** button
-
- Use the UP/DOWN ARROW buttons to change the SET and RESET priorities as needed
 - Click the **NEXT** button



- Type in the Point you wish to use.
If you need to, use the **TAGLIST** button to find the point(s)
- Click the **FINISH** button

This would be a typical horn circuit output:

```
[INPUTS]
[END INPUTS]
[OUTPUTS]
  HOLD ON PRI=99
    SIG3 | SIGB | SIGNAL | TYPICAL SMOKE DETECTOR
[END OUTPUTS]
```



When you put it all together it will look like this.

```
[INPUTS]
  STATUS DETECT
    M1-1-0 | PHOTO | SMOKE | TYPICAL SMOKE DETECTOR
[END INPUTS]
[OUTPUTS]
  HOLD ON PRI=99
    SIG3 | SIGB | SIGNAL | TYPICAL SMOKE DETECTOR
[END OUTPUTS]
```

NOTE: SIG3 must be turned off through either another Custom Control Equation OR by placing it into L73 (Control Points Off On Rest) OR L74 (Control Points Off On Signal Silence). This would be application dependant.


Error Checking

The Custom Control Editor will automatically check for valid data in an equation.

Invalid Data

When points are tagged for input/output operations, the CCE will check the validity of the point for the specific operation. If the tagged point is invalid, an error message is displayed.

This error message states that you must make a change in your custom control equation. It will not function as configured.



Syntax Error

Syntax errors occur when statements are misplaced or misspelled within a CCE equation. The editor does not allow statements to be written out of order but this can occur during copy and paste or direct statement editing.

If a does occur it will be highlighted with red text in the statement.

[INPUTS]

STATUS ON

A1 | ANALOG | ANALOG | NUMBER OF SYSTEM SUPERVISORIES

AND NOT STATUS ON

P34 | DIGITAL | TROUBLE | CITY DISCONNECT

[END INPUTS]

[OUTPUTS]

TRACK ON PRI=9,9

* | N/A | NONE |

L1162 | LIST | OUTPUT | CITY SUPERVISORY RELAY POINTS

[END OUTPUTS]


In this example a list that doesn't exist was selected. The way to correct this error is to edit the statement above and select the correct point list.

Input Logic Operators

By moving the cursor bar to the [INPUTS] line of a CCE equation and selecting the Add radio button the Inputs List can be displayed. The window that pops up is menu list contains the logic operations (Opcodes) that can be performed on an input point.


The menu options are as follows:

1. Point Status
2. Continue current input state
3. Recall memory register
4. Any N points in a list
5. Compare Day-of-Week value
6. Compare analog/heat sensor



Some of the logic operators can only be used as a second input. For example, a DELAY will begin only when the previous input statements are evaluated true. For this reason the SAVE, DELAY, CYCLE and COUNT options are not found in the initial inputs list.

- Complement current input state
- Save state to memory register
- Delay output actions
- Cycle output actions
- Count inputs




This is a simple example:

```
[INPUTS]
STATUS FIRE
L18 | LIST | MIXED | GENERAL FIRE ALARM MONITOR ZONES (this is your first input)
DELAY CONSTANT 5 (this second input is now allowed because you have a first input)
A264 | Analog Timer | ANALOG | TIMER | CUSTOM CONTROL - TIMER
[END INPUTS]
[OUTPUTS]
[END OUTPUTS]
```

Each input has a point qualifier that defines the state of an input being used.

- AND
- OR
- AND NOT
- OR NOT



OR Status Detect 1 or more points


Choosing this option will eliminate the need for selecting a qualifier. The logic OR is performed on the selected points for the Detect qualifier.

If there is only one input point the equation will read:

```
[INPUTS]
  Status Detect
    <point specifier>
[[END INPUTS]]
```

If there is more than one input point the equation will read:

```
[INPUTS]
  Status Detect
    <point specifier>
  OR Status Detect
    <point specifier>
[[END INPUTS]]
```



Where:


Point specifier is a valid selection made from the **Tag List**.

Example:

```
[INPUTS]
  STATUS DETECT
  M1-1-0 | PHOTO | SMOKE | ZONE 1 ALARM
  OR STATUS DETECT
  M1-5-0 | PHOTO | SMOKE | ZONE 2 ALARM
[END INPUTS]
[OUTPUTS]
  HOLD ON PRI=9,9
  AUX3 | RELAY | RELAY | DAMPER CONTROL RELAY
[END OUTPUTS]
```

The input would go TRUE on the leading edge of the Zone 1 Alarm OR on the leading edge of the Zone 2 Alarm. In order for this equation to work correctly, AUX3 would need to be held off in another equation or placed into list 73 or 74 (Off on SYSTEM RESET or ALARM SILENCE)

Ladder Logic:



OR 1 or more points

Choose this option to perform a logic OR on the selected input points. Any of the valid qualifiers can be selected from the Qualifiers List.

If there is only one input point the equation will need:

```
[INPUTS]
    Status <qualifier>
        <point specifier>
[[END INPUTS]]
```

If there is more than one input point the equation will read:

```
[INPUTS]
    Status <qualifier>
        <point specifier>
    OR Status <qualifier>
        <point specifier>
[[END INPUTS]]
```

Where:

Qualifier is a selection made from the **Qualifiers List**.

Point specifier is a selection made from the **Tag List**.




Example:

```
[INPUTS]
    STATUS ON
        P512 | DIGITAL | UTILITY | USER PSEUDO
    OR STATUS FIRE
        M1-5-0 | PHOTO | SMOKE | ZONE 2 ALARM
[[END INPUTS]]
```

Qualifiers Include:

Each qualifier has a specific operation that's geared towards the function of the input point. Not all qualifiers will work with all point types.

ON	NORM	CHL2	RAW SHORT	PRI2ALARM
FIRE	SHORT DETECT	CHL3	CONTROL ON	DETECT
DISABLE	OPEN DETECT	CHL4	CONTROL OFF	LEVEL
TROUBLE	ABNORM DET	ACK	DIRTY	AM THRESHOLD
TBLACK	DETECT	DOWN	DATA N/A	
McCulloh TBL	DISCONNECT	UP	SMOKE LEVEL	
FIRE ALARM ACK	EMPTY	CENTER	DISARMED	
SHORT	SUPV	RAW NORMAL	ACTIVE	
OPEN	SUPVACK	RAW ABNORM	PRIORITY 2 ALARM	
ABNORM	CHL1	RAW OPEN	PRI2 ALARM ACK	



OR NOT Status 1 or more points

This option is selected to perform a logic inverse (NOT) of a single point specifier. If more than one point is selected it will perform the logic inverse of several points OR-ed. Any of the valid qualifiers can be selected from the Qualifiers List.

Key points to remember:

- The input is TRUE if the point specifier is NOT in the state of the qualifier.
 Example: If a pseudo point needs to be turned on when a Photo Smoke is not in the FIRE state, the input would be:

```

[INPUTS]
  NOT STATUS FIRE
  M1-1-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
[END INPUTS]
```
- When **Trouble Shooting:**
 As a rule, points are off on system startup or a warm start. If a control point (relay/signal) turns on unexpectedly during system startup or warm start, look to see if an OR NOT Status<qualifier> is being used improperly.

If there is only one input point the equation will need:

```

[INPUTS]
  NOT Status <qualifier>
  <point specifier>
[END-INPUTS]
```

If there is more than one input point the equation will read:

```

[INPUTS]
  NOT Status <qualifier>
  <point specifier>
  OR NOT Status <qualifier>
  <point specifier>
[END-INPUTS]
```

Where:

Qualifier is a selection made from the **Qualifiers List**.

Point specifier is a selection made from the **Tag List**.

Example: (As long as M1-1 is not ON the input will be true)

```

[INPUTS]
  NOT STATUS ON
  M1-1-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
[END INPUTS]
[OUTPUTS]
  TRACK ON PRI=9,9
  AUX3 | RELAY | RELAY | ELEVATOR RELAY
[END OUTPUTS]
```

Ladder Logic:



AND 1 or more points

Choose this option to perform a logic AND on the selected input points. Any of the valid qualifiers can be selected from the Qualifiers List.

Key points to remember:

- The AND input qualifier states that both inputs must be TRUE for the output of the equation to work.
- If there is only one input point the equation is identical to the OR 1 or more points option.
- You **cannot** AND STATUS DETECT two points in an equation – The odds of having two points coming in at the exact same time is highly unlikely (DETECT watches the leading edge and only lasts one polling cycle of the processor).

If there is only one input point the equation will need:

```
[INPUTS]
  Status <qualifier>
    <point specifier>
[[END INPUTS]]
```

If there is more than one input point the equation will read:

```
[INPUTS]
  Status <qualifier>
    <point specifier>
  AND Status <qualifier>
    <point specifier>
[[END INPUTS]]
```

Where:


Qualifier is a selection made from the **Qualifiers List**.

Point specifier is a selection made from the **Tag List**.

Example:

```
[INPUTS]
  STATUS ON
  M1-1-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
  AND STATUS ON
  M1-5-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
[END INPUTS]
[OUTPUTS]
  TRACK ON PRI=9,9
  AUX3 | RELAY | RELAY | ELEVATOR RELAY
[END OUTPUTS]
```

Ladder Logic:



AND NOT 1 or more points

Choose this option to perform the logic inverse (NOT) of a group of input points ANDed. Any of the valid qualifiers can be selected from the Qualifiers List.

Key points to remember:

- Used particularly for Bypass Equations, but has many other useful purposes
- If there is only one input point the equation is identical to the OR NOT 1 or more points option.

If there is only one input point the equation will need:

```
[INPUTS]
  Not Status <qualifier>
    <point specifier>
[[END INPUTS]]
```

If there is more than one input point the equation will read:

```
[INPUTS]
  NOT Status <qualifier>
    <point specifier>
  AND NOT Status <qualifier>
    <point specifier>
[[END INPUTS]]
```

Where:

Qualifier is a selection made from the **Qualifiers List**.


Point specifier is a selection made from the **Tag List**.

Example:

```
[INPUTS]
  STATUS DETECT
  L18 | LIST | ASENSOR | GENERAL FIRE ALARM MONITOR ZONES
  AND NOT STATUS ON
  P512 | DIGITAL | TROUBLE | BYPASS PSEUDO
[END INPUTS]
[OUTPUTS]
  TRACK ON PRI=9,9
  AUX3 | RELAY | RELAY | TYPICAL RELAY
[END OUTPUTS]
```

In this example, the AND NOT input qualifier prevents AUX3 from turning on if P512 is on. Generally, a bypass pseudo like P512 would be associated to a button on the front panel display (TOF P512 for the switch, and ON P512 on the LED)

Ladder Logic:



CONTINUE current input state

The CONTINUE instruction on the input side of an equation acts as a reminder to the programmer that the previous SMPL input condition result is being used in the current equation.

Key points to remember:

- If you are writing 2 equations where the input of the previous equation will also be required as part of the input of the next equation the CONTINUE Opcode could save you some times.
- The rule exists, however, that you must use the CONTINUE Opcode on the next consecutive equation.

Example:

Equation 1

```
[INPUTS]
  STATUS FIRE
  M1-1-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
  AND STATUS FIRE
  M1-2-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
[END INPUTS]
[OUTPUTS]
  HOLD ON PRI=9,9
  AUX3 | RELAY | RELAY | TYPICAL CONTROLLING RELAY
[END OUTPUTS]
```

Equation 2 (Next consecutive equation)

```
[INPUTS]
  CONTINUE
  AND NOT STATUS ON
  P512 | DIGITAL | TROUBLE | SIGNAL BYPASS PSEUDO
[END INPUTS]
[OUTPUTS]
  HOLD ON PRI=9,9
  SIG3 | SIGB | SIGNAL | TYPICAL HORN CKT
[END OUTPUTS]
```

If the input of the first equation is evaluated as true, the CONTINUE of the second equation is evaluated as true. So, the STATUS FIRE of M1-1-0 and M1-2-0 will be required in the next consecutive equation also.

The CONTINUE instruction is useful if a long complicated equation is followed by a similar equation (timesaver).

COMPLEMENT current input state

The COMPLEMENT instruction works the same as the logic inverse (NOT). The difference is that it will invert the entire input condition, rather than a single operand. This can be a tricky opcode.

Example:

Example: 1 (Good equation)

```
[INPUTS]
  STATUS FIRE
  M1-1-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
  AND STATUS FIRE
  M1-2-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
  AND STATUS FIRE
  M1-3-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
  AND STATUS ON
  P512 | DIGITAL | TROUBLE | SIGNAL BYPASS PSEUDO
  COMPLEMENT
[END INPUTS]
[OUTPUTS]
  HOLD ON PRI=9,9
  SIG3 | SIGB | SIGNAL | TYPICAL HORN CKT
[END OUTPUTS]
```

As long as the logic condition prior to the COMPLEMENT instruction is FALSE, the entire input is TRUE.

Example: 2 (Poor equation)

```
[INPUTS]
  STATUS FIRE
  M1-4-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
  AND STATUS FIRE
  M1-6-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
  AND STATUS FIRE
  M1-7-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
  AND STATUS FIRE
  M1-8-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
  AND STATUS FIRE
  M1-9-0 | PHOTO | SMOKE | ANY TYPICAL SMOKE DETECTOR
  COMPLEMENT
[END INPUTS]
[OUTPUTS]
  HOLD ON PRI=9,9
  SIG4 | SIGB | SIGNAL | TYPICAL HORN CKT
[END OUTPUTS]
```

SIG4 is going to turn ON when the panel is powered up or warm started or should one of the smoke detectors go into alarm out of the group and be reset. This is probably not a desired operation.

SAVE state to memory register **&** **RECALL memory register**

The SAVE instruction is used to save the current input condition (true/false) to a memory register. There are 8 memory registers. This saved input condition can then be RECALL-ed anywhere in the current program.

Key points to remember:

- An entire input or a partial input can be saved to a register. When a partial input is saved, the logic condition of the elements preceding the SAVE instruction is saved to the register.
- This is a good instruction to use when a **multi**-statement input is being used on multiple equations. It saves repetitious writing of the same input.
- It works on the line of the CONTINUE opcode, the only difference being, you can “RECALL” the input anywhere in the program.
- It does not require the next consecutive equation to require the input as “CONTINUE” would.
- The SAVE-d input condition is not part of the current equation's input logic condition.
- The RECALL instruction is used to track the SAVE-d input condition.
- Careful care needs to be kept when dealing with registers, when recalling a register, keep close attention to when it was last saved to. This can be critical.

The analog A6 is a pseudo which indicates the current hour (military time). The analog A7 is a pseudo which indicates the current minute.

The intent of these equations was to turn on AUX3 at 7:30 am and at 9:00 pm.

Example 1: (Good equation)

```
[INPUTS]
  IF ANALOG = CONSTANT 7.000000 CNTS
    A6 | Compare Analog | ANALOG | ANALOG | CURRENT HOUR
  AND IF ANALOG = CONSTANT 30.000000 CNTS
    A7 | Compare Analog | ANALOG | ANALOG | CURRENT MINUTE
  SAVE 1
  IF ANALOG = CONSTANT 21.000000 CNTS
    A6 | Compare Analog | ANALOG | ANALOG | CURRENT HOUR
  AND IF ANALOG = CONSTANT 0.000000 CNTS
    A7 | Compare Analog | ANALOG | ANALOG | CURRENT MINUTE
  OR RECALL 1
[END INPUTS]
[OUTPUTS]
  TRACK ON PRI=9,9
    AUX3 | RELAY | RELAY | TIME CONTROL RELAY
[END OUTPUTS]
```

If the equation in Example 1 is written out long hand, the SAVE and RECALL is equivalent to putting parenthesis around the input prior to the SAVE instruction and OR-ing it with the input prior to the RECALL:


When (A6=21 AND A7=0) OR (A6=7 AND A7=30), TRACK AUX3 ON.

It's important to note that the input prior to the SAVE is not part of the input's logic condition. The RECALL is the element that is evaluated. If there was no RECALL, the logic condition of the input would be equivalent to:

INPUTS:

When analog: A6 is equal to value: 21
 AND When analog: A7 is equal to value: 0

Ladder Logic: (from Example 1)




Note: It is important to RECALL from the correct register. You can SAVE to registers 1-8, and you can RECALL from the same choices. In this case, recall from the same register (1) that it was SAVE-d to.

Example 2: (Poor equation)

```
[INPUTS]
IF ANALOG = CONSTANT 7.000000 CNTS
  A6 | Compare Analog | ANALOG | ANALOG | CURRENT HOUR
AND IF ANALOG = CONSTANT 30.000000 CNTS
  A7 | Compare Analog | ANALOG | ANALOG | CURRENT MINUTE
OR IF ANALOG = CONSTANT 21.000000 CNTS
  A6 | Compare Analog | ANALOG | ANALOG | CURRENT HOUR
AND IF ANALOG = CONSTANT 0.000000 CNTS
  A7 | Compare Analog | ANALOG | ANALOG | CURRENT MINUTE
[END INPUTS]
[OUTPUTS]
TRACK ON PRI=9,9
  AUX3 | RELAY | RELAY | TIME CONTROL RELAY
[END OUTPUTS]
```

If the equation in Example 2 is written out long hand, the equation reads:
 When A6=7 AND A7=30 OR A6=21 AND A7=0, TRACK AUX3 ON.

Ladder Logic: (from Example 2)



The Ladder Logic shows that the output will never be turned on. A7=30 and A7=0 would both have to be TRUE for **AUX3** to turn on. To get the correct logic, the SAVE and RECALL must be used.

DELAY output actions

This instruction will delay the activation of the output when the input goes true. There are two additional components to the delay equation, the setpoint and the analog running timer. The setpoint can be entered as a constant value or as an analog pseudo which contains the setpoint value.

Key points to remember:

- The advantage to using an analog pseudo which contains the set point value is the value can be changed from the front panel display if different delay times would need to be tested. This can be useful for testing purposes.
- By using a constant value another download would be required to test a different time.
- When the input elements preceding the DELAY go true, the running timer will begin counting from zero up to the value of the setpoint.
- The analog running timer is initially off. When the timer reaches the value of the setpoint, it turns on, making the entire input true.
- The analog running timer remains on until the input goes false.
- If the input elements preceding the DELAY go from true to false before the running timer reaches the setpoint value. The DELAY stops and the running timer resets to 0.
- **DO NOT USE A DETECT** on the input prior to the DELAY. The delay is only active while the input is true. A detect is only true for one poll cycle.
- The CCE will prompt you with the next available analog pseudo, when selecting a setpoint analog or an analog running timer. You can select the next available pseudo (the first user analog pseudo that is not labeled) or enter an alternate. If the pseudo is not labeled it will be given a generic label, CUSTOM CONTROL - ANALOG or CUSTOM CONTROL - TIMER.

Example 1: (Setpoint taken from analog)

Equation 1: The Setpoint must be SET via Custom Control

```
[INPUTS]
STATUS ON
A34 | ANALOG | TIMER | SYSTEM STARTUP PULSE TIMER
[END INPUTS]
[OUTPUTS]
SET CONSTANT 10 CNTS
A260 | ANALOG | ANALOG | ANALOG TIMER SET POINT VALUE OF 10
[END OUTPUTS]
```

Equation 2:

```
[INPUTS]
STATUS FIRE
M1-6-0 | PHOTO | SMOKE | ELEVATOR LOBBY 3
OR STATUS FIRE
M1-7-0 | PHOTO | SMOKE | ELEVATOR LOBBY 4
DELAY ANALOG
A261 | Analog Timer | ANALOG | TIMER | CUSTOM CONTROL - TIMER
A260 | Analog Timer SetPoint | ANALOG | TIMER | CUSTOM CONTROL - TIMER
[END INPUTS]
[OUTPUTS]
HOLD ON PRI=9,9
M1-2-0 | SIGIAM | SIGNAL | ELEVATOR STROBE
[END OUTPUTS]
```


Example 2: (Constant value setpoint)

```
[INPUTS]
STATUS FIRE
M1-6-0 | PHOTO | SMOKE | ELEVATOR LOBBY 3
OR STATUS FIRE
M1-7-0 | PHOTO | SMOKE | ELEVATOR LOBBY 4
DELAY CONSTANT 30
A263 | Analog Timer | ANALOG | TIMER | CUSTOM CONTROL - TIMER
[END INPUTS]
[OUTPUTS]
HOLD ON PRI=9,9
M1-2-0 | SIGIAM | SIGNAL | ELEVATOR STROBE
[END OUTPUTS]
```

In these examples:

- The timer will begin running when M1-6 or M1-7 are in the FIRE state.
- The timer will run until it reaches the value of the setpoint.
- When the timer reaches the value of the setpoint the timer turns on, making the input true.
- The signal will then turn on.

Ladder Logic:



CYCLE OUTPUT ACTIONS

This instruction will toggle the true/false condition of the input at the intervals selected. There are three additional components to the CYCLE command that are selected from CCE menus.

1. The on time setpoint:

It can be entered as a constant value or as an analog pseudo that contains the setpoint value.

2. The off time setpoint:

It can be entered as a constant value or as an analog pseudo that contains the setpoint value. It must be of the same type as the onetime setpoint, either constant or analog pseudo.

The advantage of using an analog pseudo as the setpoint is that its value can be changed from the front panel keypad. This is sometimes useful for testing purposes.

The CCE will prompt you with the next available analog pseudo, when selecting a setpoint analog or an analog running timer. You can select the next available pseudo (the first user analog pseudo that is not labeled) or enter an alternate. If the pseudo is not labeled it will be given a generic label, CUSTOM CONTROL - ANALOG or CUSTOM CONTROL - TIMER.

3. The analog running timer:

It holds the current value of the on/off time.

When the input elements preceding the CYCLE command go true, the running timer will begin counting from zero up to the value of the on time setpoint. The analog running timer is initially ON, making the input true. When the timer reaches the value of the on time setpoint, it turns OFF, making the entire input false. The analog running timer is then set to the value of the off time setpoint and counts down to zero. When the running timer reaches zero it turns ON, making the input true, and counts up to the value of the on time setpoint. This cycle will continue until the input elements preceding the CYCLE command go false, at which time the running timer is set to zero and turned OFF.

AHU Staggered Start Program is excellent example of the uses of the CYCLE opcode. – See Chapter 3 for an example of the AHU program

Example 1: (Setpoint taken from analog)

Equation 1: The Setpoints must be SET via Custom Control

```
[INPUTS]
STATUS ON
A34 | ANALOG | TIMER | SYSTEM STARTUP PULSE TIMER
[END INPUTS]
[OUTPUTS]
SET CONSTANT 5 CNTS
A265 | ANALOG | ANALOG | ANALOG TIMER SET POINT
SET CONSTANT 295 CNTS
A266 | ANALOG | ANALOG | ANALOG TIMER SET POINT
[END OUTPUTS]
```

Equation 2:

```
[INPUTS]
  STATUS ON
  A1 | ANALOG | ANALOG | NUMBER OF SYSTEM SUPERVISORIES
  CYCLE ANALOG
  A267 | Analog Timer | ANALOG | TIMER | CUSTOM CONTROL - TIMER
  A265 | Analog ON Timer SetPoint | ANALOG | TIMER | CUSTOM CONTROL - TIMER
  A266 | Analog OFF Timer SetPoint | ANALOG | TIMER | CUSTOM CONTROL - TIMER
[END INPUTS]
[OUTPUTS]
  TRACK ON PRI=9,9
  0-7-2 | MCODE7 | PIEZO | Master Output - Display Sonalert
[END OUTPUTS]
```


Example 2: (Constant value setpoint)

```
[INPUTS]
  STATUS ON
  A1 | ANALOG | ANALOG | NUMBER OF SYSTEM SUPERVISORIES
  CYCLE CONSTANT 5 295
  A267 | Analog Timer | ANALOG | TIMER | CUSTOM CONTROL - TIMER
[END INPUTS]
[OUTPUTS]
  TRACK ON PRI=9,9
  0-7-2 | MCODE7 | PIEZO | Master Output - Display Sonalert
[END OUTPUTS]
```

In these examples:

- If a supervisory comes in, A1 will turn ON. A1 is a system pseudo which turns ON when any supervisory condition exists in the system.
- A1 turning ON will make the input preceding the CYCLE command true. This will start the cycle. The analog running timer (A267) will turn ON and count from 0 to 5 seconds (A265 in Example12). While A267 is ON, the entire input is true causing the output action to take place. In this case the Master Piezo will turn ON.
- When A267 reaches 5, it turns OFF and is set to 295 seconds (A266 in Example 1). A267 turning OFF will make the entire input false, turning the piezo OFF. A267 remains OFF as it counts down to 0.
- When A267 reaches zero it turns ON, making the input true and activating the piezo. The analog running timer (A267) will then count from 0 to 5 seconds. This cycle will continue until there are no more supervisories in the system and A1 turns OFF.
- When A1 turns OFF, A261 will turn OFF and its value is set to 0.

Ladder Logic:



COUNT inputs

This instruction serves as a digital counter. There are three additional components to the COUNT command which are selected from CCE menus.

1. The setpoint.

The setpoint can be a constant value or an analog pseudo that contains the setpoint value.

2. The analog running counter.

The analog running counter will initially have a value of zero. Each time the logic condition of the elements preceding the COUNT command goes true, the counter is incremented by one until it reaches the value of the setpoint. Any additional off to on transitions after the count has been reached will cause the input to go true for one poll cycle.

3. The reset point specifier.

When the reset point goes true the count is reset to zero.

Key points to remember:

- Every time the input goes true the counter will increment
- **DON'T USE TRACK FOR THE OUTPUT.** The input is only true for one poll cycle, therefore a TRACK would turn the output on for only one poll cycle.

See Chapter 3 for an example of the uses of COUNT inputs.

Example 1: (Setpoint taken from analog)

Equation 1: The Setpoints must be SET via Custom Control

```
[INPUTS]
STATUS ON
A34 | ANALOG | TIMER | SYSTEM STARTUP PULSE TIMER
[END INPUTS]
[OUTPUTS]
SET CONSTANT 10 CNTS
A440 | Set-Analog (Destination) | ANALOG | ANALOG | THIS IS THE SET POINT VALUE IS 10
[END OUTPUTS]
```

Equation 2:

```
[INPUTS]
STATUS ON
P530 | DIGITAL | UTILITY | THIS PSEUDO WILL INCREMENT COUNTER A600
COUNT ANALOG RESET ON
A441 | Analog Counter | ANALOG | ANALOG | WILL INCREMENT ON TOGGLE OF INPUT
P535 | Reset-Point | DIGITAL | UTILITY | THIS IS THE RESET PSEUDO
A440 | Analog Counter SetPoint | ANALOG | ANALOG | THIS IS THE SET POINT VALUE IS 10
[END INPUTS]
[OUTPUTS]
[END OUTPUTS]
```

A440 is first SET during Start up. A441 will increment when the input is being turned on. It will maintain that count until P530 turns off and on again or if P535 turns on. At anytime during the count sequence, if P535 turns on, A441 will return to a value of 0 (reset).

Now, A441 can be used in equations like:

```
[INPUTS]
IF ANALOG = CONSTANT 1.000000 CNTS
A441 | Compare Analog | ANALOG | ANALOG | WILL INCREMENT ON TOGGLE OF INPUT
```

```

[END INPUTS]
[OUTPUTS]
  TRACK ON PRI=9,9
    P531 | DIGITAL | UTILITY | TURN ON WHEN COUNTER = 1
[END OUTPUTS]

```


Example 2: (Constant value setpoint)

```

[INPUTS]
  STATUS ON
    P513 | DIGITAL | UTILITY |
  COUNT CONSTANT 10 RESET ON
    A268 | Analog Counter | ANALOG | COUNTER | CUSTOM CONTROL - COUNTER
    P514 | Reset-Point | DIGITAL | UTILITY |
[END INPUTS]
[OUTPUTS]
  HOLD ON PRI=9,9
    AUX3 | RELAY | RELAY | ELEVATOR RELAY
[END OUTPUTS]

```

In this example, each time P513 turns on the running counter (A268) is incremented by one. Any “off” to “on” transitions after A268 is equal to 10 will cause the input to go true for one poll cycle. In other words, the 11th off to on transition of P513 will make the input true. Any addition “off” to “on” transitions will make the input true. If P514 turns on, A268 is set to zero.

Ladder Logic (Example 2):

ANY n points in a point list

This special input is used in applications where a specific number of points in a group must be true for any output to occur. The input will go true when the selected number of points go true in the selected list.

Key points to remember:

- This special input is always used with a list, either user created or system.
- The inputs being counted must be steady.
- When **Trouble Shooting:**

If the the Any n points in a point list instruction is not working it is probably because you are using a DETECT on the input. **DO NOT USE A DETECT** on the input prior to the Any n points in a point list instruction. The the Any n points in a point list instruction is only active while the input is true. A detect is only true for one poll cycle.

Example:

```
[INPUTS]
  ANY 2 FIRE
  L19 | LIST | ASENSOR | ELEVATOR SHAFT 1 PRIMARY MONITOR ZONES
  AND NOT ANY 2 FIRE
  L20 | LIST | EMPTY | ELEVATOR SHAFT 1 ALTERNATE MONITOR ZONES
[END INPUTS]
[OUTPUTS]
  TRACK ON PRI=8, 8
  L21 | LIST | EMPTY | ELEVATOR SHAFT 1 PRIMARY RELAY
[END OUTPUTS]
```

This equation will activate the primary elevator capture relays only if two lobby smokes come into alarm. The output priority of 8 will override the Default SMPL.

Compare analog

This Special input is used to compare the value of one analog pseudo to another value, either analog pseudo or a constant.

The six relational operators are:

- equal to
- not equal to
- greater than
- greater than or equal to
- less than
- less than or equal to

Key points to remember:

- This Special input is usually followed by a “setup equation”, see example

Example:

Equation 1: This equation will get Analog Timer A261 to pulse for 10 – Setup Equation

```
[INPUTS]
  STATUS FIRE
  M1-1-0 | PHOTO | SMOKE | SMOKE DETECTOR 1
[END INPUTS]
[OUTPUTS]
  PULSE CONSTANT 10
  A261 | Analog Timer | ANALOG | TIMER | CUSTOM CONTROL - TIMER
[END OUTPUTS]
```

Equation 2:

```
[INPUTS]
  IF ANALOG = CONSTANT 5.000000 CNTS
  A261 | Compare Analog | ANALOG | TIMER | CUSTOM CONTROL - TIMER
[END INPUTS]
[OUTPUTS]
  HOLD ON PRI=9,9
  SIG4 | SIGB | SIGNAL | DELAYED HORN CKT
[END OUTPUTS]
```

This equation will hold SIG4 on when A261 reaches a value of 5. SIG4 would be placed in L74 to turn off on pressing the Alarm Silence key or L73 to turn off when the System Reset key is pressed.

Compare DOW value

This Special input is used in applications where operations are dependent on the day of week. A single day or combinations of the seven days of the week can be selected from a CCE menu list.

This application is often used when door control relay is day and time sensitive. The hours are based on military time.

Example:

```
[INPUTS]
  WHEN DAY IS MON, TUE, WED
  AND IF ANALOG = CONSTANT 5.000000 CNTS
    A6 | Compare Analog | ANALOG | ANALOG | CURRENT HOUR
  AND IF ANALOG = CONSTANT 30.000000 CNTS
    A7 | Compare Analog | ANALOG | ANALOG | CURRENT MINUTE
[END INPUTS]
[OUTPUTS]
  HOLD OFF PRI=6,9
  AUX3 | RELAY | RELAY | DOOR CONTROL RELAY
[END OUTPUTS]
```

This equation states that if it is Monday, Tuesday, Wednesday AND the current hour is 5:00 a.m. and the current minute is 30 (essentially 5:30 a.m.), then hold off AUX3 at a priority 6.

Output Logic Operators

By moving the cursor bar to the [END-OF-OUTPUTS] line of a CCE equation and striking ENTER, the Outputs List can be displayed. This menu list contains the instructions for the activation of outputs. The menu options are as follows:

- * HOLD POINT 1 or more points
- * TRACK POINT 1 or more points
- * SET POINT 1 or more points
- * PULSE ANALOG 1 or more points
- DISCONNECT MONITOR POINT 1 or more points
- DISABLE POINT 1 or more points
- LED FUNCTIONS (FAST/SLOW/ON/OFF)
- RESET Monitor or Analog points
- SET ANALOG VALUES
- PRINT MESSAGE
- * MARCH CODE CONTROL POINTS
- * PNIS CODE CONTROL POINTS
- * CALIFORNIA CODE CONTROL POINTS
- * CODE (GENERIC) CONTROL POINTS
- CHLCODE MAPNET CHANNEL
- * DEVICE CODE SMART SENSORS
- SETALARM THRESHOLD SMART SENSORS
- SET RATE OF RISE HEAT SENSORS
- DISARM POINT 1 or more points
- QCODE ON A POINT
- UVOLUME SET AUDIO VOLUME
- * UPLAY AUDIO MESSAGE (PLAY)
- * UPLAYQ AUDIO MESSAGE (PLAYQ)
- * USILENCE SILENCE AUDIO CHANNEL (SILENCE)
- * USPEAKER CIRCUITS ON/OFF (SPEAKER)
- * UPNISV VOICE CODE SPKR CRTS (PNISV)
- * UPNIST TONE CODE SPKR CRTS
- * UPNISM MESSAGE CODE SPKR CRTS (PNISM)
- * UROUTE ROUTE AUDIO (ROUTE)
- * MARCH 60 ON A POINT
- * MARCH 120 ON A POINT
- * TEMPORAL ON A POINT
- * MARCH 20 ON A POINT
- * HPULSE
- * TCODE ON A LIST OF TRUEALERT POINTS
- ACKNOWLEDGE TBL/SUP/FIRE/PRI2

* These output operators involve prioritized operation.

Operators in parenthesis are for legacy audio but are used in upgraded panels that still contain legacy controllers. Legacy commands will not be referenced here.

Output Priority

All output points, with the exception of annunciator LEDs, have an associated priority level.

Key points to remember:

- For a prioritized output operation to take place, the Output Logic Operator must act at a priority greater than or equal to the output point's present priority.
- The priority level of an output point can be from 0 to 15, with 0 being the highest priority. Most of the outputs are initially set to a priority level of 15. The exception is the digital pseudo. Some of the digital pseudos are set to a priority level of 0 by the system.
- The output operators (HOLD, TRACK, etc.) can act at priority levels from 2 to 15. Any pseudo points that are set to a priority 0 by the system cannot be affected by SMPL Output Logic Operators.
- At start-up, all points that are in a HOLD, TRACK, or SET statement (Default or User SMPL) are set to the reset priority of the equation.
- The Set/Reset Priority are dependent of the status of the input and the number value of the priority. As long as the input is true, the output will be sitting at the set priority. **Note: The output does not change, the priority does.**
 - Example: PRI=8,9 where 8 is the Set Priority and 9 is the Reset Priority.
- The output points will toggle to the Reset Priority one of two ways.
 - **Output points in a HOLD, TRACK or SET command will toggle to the reset priority when the input of the equation goes false.**
 - **Output points in a MARCH, PNIS, CALIFORNIA, CODE, UPLAY, UROUTE, USILENCE, USPEAKER, UPNISV, UPNIST, UPNISM, UVOLUME, UPNIST, MARCH20, MARCH60, MARCH120, and TEMPORAL command will toggle to the reset priority when the command is no longer active.** For the UROUTE, USILENCE and USPEAKER commands this is instantaneous. For any UPLAY type equation this is when the message is complete. For any CODE type equation this is when the code is complete.


HOLD 1 or more points

This instruction will HOLD the output point according to the option picked (**ON**, **OFF** or **PRI**) until another action (via Custom Control or System Lists) affects the output point.

This instruction will turn the output point ON when the input logic goes true. The priority 9,9 is assumed.

Key points to remember:

- When the input goes true:
 - o “**HOLD ON**” will turn the output points on if the set priority is greater than or equal to the output points current priority. When the input goes false the output will remain on, but the output point's priority will toggle to the reset priority level. The output point will remain ON until either:
 - The HOLD OFF command is issued.
 - OR**
 - When Alarm Silence or System Reset is activated, if the point is put into L73 or L74.
 - o “**HOLD OFF**” will turn the output points off if the set priority is greater than or equal to the output points current priority. When the input goes false the output will remain off, but the output point's priority will toggle to the reset priority level.
 - o “**HOLD PRI**” will not affect the output points on/off condition. The purpose of this instruction is to change the priority level of the output points. In order for this instruction to take effect the set priority must be greater than or equal to the output point's current priority. When the input goes true the output point's priority will toggle to the set priority. When the input goes false, the output point's priority will toggle to the reset priority.



The output of the equation will read:

```
[OUTPUTS]
  HOLD <qualifier>PRI==<sp>, <rp>
    <output point>
[END OUTPUTS]
```

Where:

- Output point** is selected from the **Tag List**.
- Point specifier** is a selection made from the **Tag List**.
- SP** is the set priority selected from the choice list.
- RP** is the reset priority selected from the choice list.

The valid output points for HOLD are:

SIG, AUX, DIGITAL Pseudo, LIST Pseudo, MASTER Slot 7, IO Output, MAPNET or IDnet Output

Example:

```
[INPUTS]
  STATUS DETECT
  M1-9-0 | PHOTO | SMOKE | ELEVATOR LOBBY 6
  OR STATUS DETECT
  ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
  HOLD ON PRI=9,9
  SIG3 | SIGB | SIGNAL | SIGNAL CARD 1 CIRCUIT SIG3
  HOLD OFF PRI=9,9
  AUX3 | RELAY | RELAY | TYPICAL RELAY
[END OUTPUTS]
```

In this example, when the input goes true SIG3 and AUX3 are turned ON at a priority 9. When the input goes false SIG3 and AUX3 remain on at a priority 9.

In this example, when the input goes true, SIG3 turned ON at a priority 9 and AUX3 are turned OFF at a priority 9. When the input goes false SIG3 and AUX3 remain on at a priority 9.


An output statement can have use different priorities with different devices. In the example about, if desired SIG3 could be at priority 8,9 and AUX3 at priority 9,9.

TRACK 1 or more points

This instruction will TRACK the output point according to the option picked (**ON, OFF or PRI**) while the input is TRUE

Key points to remember:

- When the input goes true:
 - o “**TRACK ON**” will turn the output points on if the set priority is greater than or equal to the output points current priority. When the input goes false the output will turn off and the output point's priority will toggle to the reset priority level.
 - o “**TRACK OFF**” will turn the output points off if the set priority is greater than or equal to the output points current priority. When the input goes false the output will turn on and the output points priority will toggle to the reset priority level. It's important to remember that if the input is normally false, **THE OUTPUT WILL ALWAYS BE ON** until the input goes true which could cause devices to cycle on during a system start up.
 - o “**TRACK PRI**” will not affect the output points on/off condition. The purpose of this instruction is to change the priority level of the output points. In order for this instruction to take effect the set priority must be greater than or equal to the output points current priority. When the input goes true the output point's priority will toggle to the set priority. When the input goes false the output point's priority will toggle to the reset priority.
- When **Trouble Shooting:**
 - o Even though this is a valid and available output, TRACK OFF is not recommended. A major inherent issue with TRACK OFF is that any point that is tracked off will turn ON, on system start up. This causes a lot of issues.



The output of the equation will read:

```
[OUTPUTS]
  TRACK <qualifier>PRI==<sp>, <rp>
    <output point>
[END OUTPUTS]
```

Where:

Output point is selected from the **Tag List**.

Point specifier is a selection made from the **Tag List**.

SP is the set priority selected from the choice list.

RP is the reset priority selected from the choice list.

The valid output points for TRACK are:

SIG, AUX, DIGITAL Pseudo, LIST Pseudo, MASTER Slot 7, IO Output, MAPNET or IDnet Output

Example:

```
[INPUTS]
  STATUS FIRE
  ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
  TRACK ON PRI=6,9
  SIG3 | SIGB | SIGNAL | SIGNAL CARD 1 CIRCUIT SIG3
  TRACK ON PRI=6,9
  AUX3 | RELAY | RELAY | TYPICAL RELAY
[END OUTPUTS]
```

In this example, when the input goes true SIG3 and AUX3 are turned ON at a priority 6.

When the input goes false SIG1 and AUX1 are turned off. The priority levels of SIG1 and AUX1 would toggle from a 6 to a 9.

SET 1 or more points

This instruction will SET the output point according to the option picked (**ON, OFF or PRI**) while the input is TRUE.

Key points to remember:

- The ON, OFF, and PRI options for the SET command and the action of the set and reset priorities are very similar to the TRACK command. The difference is as follows:
 - o The **TRACK** command will activate the output action on the leading edge of the input's off to on transition and will deactivate the output action on the trailing edge of the input's on to off transition.
 - o The **SET** command will activate the output action on the leading edge of the input's off to on transition and will continually refresh the output condition for each poll cycle.

The output of the equation will read:

```
[OUTPUTS]
  SET <qualifier>PRI==<sp>, <rp>
    <output point>
[END OUTPUTS]
```

Where:

Output point is selected from the **Tag List**.

Point specifier is a selection made from the **Tag List**.

SP is the set priority selected from the choice list.

RP is the reset priority selected from the choice list.

The valid output points for SET are:

SIG, AUX, DIGITAL Pseudo, LIST Pseudo, MASTER Slot 7, IO Output, MAPNET or IDnet Output

Example:

```
Equation 1:
[INPUTS]
  STATUS FIRE
  ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
  SET ON PRI=6,9
  AUX3 | RELAY | RELAY | ELEVATOR RELAY
[END OUTPUTS]
```

```
Equation 2:
[INPUTS]
  STATUS ON
  P515 | DIGITAL | UTILITY |
[END INPUTS]
[OUTPUTS]
  HOLD OFF PRI=6,9
  AUX3 | RELAY | RELAY | ELEVATOR RELAY
[END OUTPUTS]
```

In this example, when the input of Equation 1 goes TRUE, AUX1 is turned ON at a priority 6. When the input of Equation 2 goes TRUE, AUX3 is turned OFF. On the next poll cycle, if the input to Equation 1 is still TRUE, AUX1 will turn back on at a priority 6. Because HOLD will only activate the output on the leading edge of the off to on transition, Equation 2 would not turn AUX3 back off until P515 is turned off and back on.

PULSE analog

This output instruction will turn ON an analog pseudo for the duration selected. The duration is determined either by a constant value or an analog value.

Key points to remember:

- The advantage of using an analog pseudo as the setpoint is that its value can be changed from the front panel keypad. This is sometimes useful for testing purposes.
- The analog running timer holds the current value of the ON time.
- The CCE will prompt you with the next available analog pseudo, when selecting a setpoint analog or an analog running timer. You can select the next available pseudo (the first user analog pseudo that is not labeled) or enter an alternate. If the analog is not labeled it will be given a generic label, CUSTOM CONTROL - ANALOG or CUSTOM CONTROL - TIMER.
- When the input goes true, the PULSE instruction will turn ON the analog running timer. The timer will count from 0 seconds up to the value of the setpoint. When the analog timer reaches the value of the setpoint it will turn off.
- The PULSE is activated on the leading edge of the off to on transition of the input, therefore the analog timer will continue to count if the input goes false.

The output of the equation, when set with a analog value, will read:

```
[OUTPUTS]
  PULSE ANALOG
    <analog timer>
    <analog setpoint>
[END OUTPUTS]
```

The output of the equation, when set with an constant value, will read:

```
[OUTPUTS]
  PULSE CONSTANT <time>
    <analog timer>
[END OUTPUTS]
```

Where:

Output point is selected from the **Tag List**.

Analog Timer is selected by the programmer – first user analog pseudo available without a label

Analog Setpoint is selected by the programmer – next user analog pseudo available without a label

TIME is selectable by HH:MM:SS or seconds.

The valid output points for PULSE are:

ANALOG Pseudos

Example:

```
Equation 1:
[INPUTS]
  STATUS ON
  P515 | DIGITAL | UTILITY |
[END INPUTS]
[OUTPUTS]
  PULSE CONSTANT 5
  A271 | Analog Timer | ANALOG | TIMER | CUSTOM CONTROL - TIMER
[END OUTPUTS]
```

```
Equation 2:  
[INPUTS]  
  STATUS ON  
  A271 | ANALOG | TIMER | CUSTOM CONTROL - TIMER  
[END INPUTS]  
[OUTPUTS]  
  TRACK ON PRI=9,9  
  AUX3 | RELAY | RELAY | TYPICAL RELAY  
[END OUTPUTS]
```

In this example, when the input goes true A271 is turned on for 5 seconds. Because an analog pseudo is the only valid point type for a PULSE equation, equation 2 is required to pulse control points.

DISCONNECT 1 or more MONITOR points

This output instruction will DISCONNECT monitor points.

Key points to remember:

- There are two options to a DISCONNECT equation, ON and OFF.
- The **DISCONNECT instruction is only valid for monitor points**, hardwired, Mapnet or IDnet.
- The ON option will drop power to the Monitor points specified, when the input goes true.
- The OFF option will put power back on the Monitor point specified, when the input goes true.

The output of the DISCONNECT ON equation will read:

```
[OUTPUTS]
  DISC ON
    <monitor point>
[END OUTPUTS]
```

The output of the DISCONNECT OFF equation will read:

```
[OUTPUTS]
  DISC OFF
    <monitor point>
[END OUTPUTS]
```

Where:

Output point is selected from the **Tag List**.
Monitor point is selected from the **Tag List**.

The valid output points for DISCONNECT are:

ZN, LIST Pseudo, MAPNET or IDnet Input

Example:

```
Equation 1:
[INPUTS]
  STATUS UP
    IO1 | INPUT | USWITCH |
[END INPUTS]
[OUTPUTS]
  DISC ON
    ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END OUTPUTS]
```

```
Equation 2:
[INPUTS]
  STATUS DOWN
    IO1 | INPUT | USWITCH |
[END INPUTS]
[OUTPUTS]
  DISC OFF
    ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END OUTPUTS]
```

DISABLE 1 or more points

This output instruction will DISABLE the specified points via software.

Key points to remember:

- There are two options to a DISABLE equation, ON and OFF.
- The ON option will **software disable** the points specified, when the input goes true.
- The OFF option will ENABLE points specified, when the input goes true.
- On an output point, a software disable means that the point cannot be affected by SMPL or by the use of switches. The point will remain in its' present condition when disabled. If it is on it will remain on. If it is off it will remain off.
- If an input point is disabled, power is not turned off to the point. Any changes in the point's status are disregarded in SMPL or by the reference address of a switch.
- If a SMPL equation or switch tries to turn a point on, all indications will be that the point is on. For example, if SIG3 is off and is disabled, it will remain off. If a zone comes in that would normally trip SIG3, SIG3's point status would indicate that it was ON. Any LEDs tracking its' ON state would turn on, but SIG3 would remain off. If SIG3 was enabled while its' point status indicated it was ON, SIG3 would turn on immediately.

The output of the DISABLE ON equation will read:

```
[OUTPUTS]
  DISABLE ON
    <monitor point>
[END OUTPUTS]
```

The output of the DISABLE OFF equation will read:

```
[OUTPUTS]
  DISABLE OFF
    <monitor point>
[END OUTPUTS]
```

Where:

Output point is selected from the **Tag List**.

Monitor point is selected from the **Tag List**.

The valid output points for DISABLE are:

ZN, SIG, AUX, LIST Pseudo, MASTER Slot 7, IO Output, MAPNET or IDnet Input/Output

Example:

Equation 1:

[INPUTS]

STATUS UP

IO1 | INPUT | USWITCH |

[END INPUTS]

[OUTPUTS]

DISABLE ON

SIG3 | SIGB | SIGNAL | SIGNAL CARD 1 CIRCUIT SIG3

DISABLE ON

AUX3 | RELAY | RELAY | TYPICAL RELAY

[END OUTPUTS]

Equation 2:

[INPUTS]

STATUS DOWN

IO1 | INPUT | USWITCH |

[END INPUTS]

[OUTPUTS]

DISABLE OFF

SIG3 | SIGB | SIGNAL | SIGNAL CARD 1 CIRCUIT SIG3

DISABLE OFF

AUX3 | RELAY | RELAY | TYPICAL RELAY


[END OUTPUTS]

LED functions (FAST/SLOW/ON/OFF)

This instruction allows SMPL control of annunciator leds.

Key points to remember:

- The LED output should only be used on led operations that are not supported by modes.
- The LED, IO lamp, or IO piezo used as the output, **MUST BE GIVEN THE SMPL MODE** for the LED instruction to work.
- If an output point is used that does not have the SMPL mode, the equation will be flagged with an error message window stating: "Point not valid with this Opcode."
- The LED instruction is a **HOLD type instruction**. If a LED is turned on, another equation is required to turn it off. The output point will not turn off when the input goes false. A LED can be turned on or off or made to flash either slow or fast.
- Remember when turning on an LED with custom control you will need to also use custom control to turn it off.



The output of the equation will read:

```
[OUTPUTS]
    FLASH <option>:
        <output point>
[END OUTPUTS]
```

Where:

- Output point** is selected from the **Tag List**.
- Option** is a selection made from the **Tag List**.

The valid output points for HOLD are:

IO Lamp or Piezo Output, Master Slot 4 Led, 64/64 Led

Example:

```
[INPUTS]
    STATUS FIRE
    L256 | LIST | MONITOR | SPECIAL ALARM LIST
[END INPUTS]
[OUTPUTS]
    FLASH FAST
    7-2-6 | LED | LEDRED | FUI USER LED FOR SWITCH
[END OUTPUTS]
```

RESET monitor or analog points

This instruction will RESET either monitor or analog points

Key points to remember:

- The two options to the RESET instruction are Reset Detector and Reset Timer.
- Reset Detector will perform a 5 second drop in power to a monitor point. This is **NOT EQUIVALENT TO SYSTEM RESET**. The 5 second drop in power will allow a smoke head, which is locked in a current limited condition, to clear. The ALARM condition at the panel does not clear.
- Reset Timer will turn off a running timer and set it's value to zero. This can also be used on a running counter.

The output of the RESET Detector equation will read:

```
[OUTPUTS]
  RESET DETECTOR
    <monitor point>
[END OUTPUTS]
```

The output of the RESET Detector equation will read:

```
[OUTPUTS]
  RESET TIMER
    <analog point>
[END OUTPUTS]
```

Where:

Output point is selected from the **Tag List**.

Monitor Point is a selection made from the **Tag List**.

Analog Point is a selection made from the **Tag List**.

The valid output points for RESET are:

ZN, MAPNET or IDnet Input, and LIST Pseudo for Reset Detector

ANALOG Pseudo and LIST Pseudo for Reset Analog

Example:

Equation 1:

```
[INPUTS]
  STATUS UP
    IO1 | INPUT | USWITCH |
[END INPUTS]
[OUTPUTS]
  RESET DETECTOR
    ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
  RESET DETECTOR
    ZN2 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN2
[END OUTPUTS]
```

Equation 2:

```
[INPUTS]
  STATUS ON
    P0 | DIGITAL | UTILITY | SYSTEM RESET KEY
[END INPUTS]
[OUTPUTS]
  RESET TIMER
    A256 | ANALOG | TIMER | CUSTOM CONTROL - TIMER
[END OUTPUTS]
```

SET analog value

This instruction will assign a value to an analog pseudo.

Key points to remember:

- The value can be entered as a constant or the value can come from another analog pseudo.
- The **SET analog value** command is **not at all related to the SET command**.

The output of the equation, when set with a analog value, will read:

```
[OUTPUTS]
  SET ANALOG
    <analog (destination)>
    <analog (source)>
[END OUTPUTS]
```

The output of the equation, when set with an constant value, will read:

```
[OUTPUTS]
  SET CONSTANT <number> <unit>
    <analog (destination)>
[END OUTPUTS]
```

Where:

Output point is selected from the **Tag List**.

Analog (Destination) is selected by the programmer – first user analog pseudo available without a label

Analog (Source) is selected by the programmer – user chooses the analog and must remember to set it via the SET Command

Number is entered in the data entry field

Unit is selected from the **Tag List**

The valid output points for HOLD are:

ANALOG Pseudo and LIST Pseudo

Example:

```
[INPUTS]
  STATUS ON
    A34 | ANALOG | TIMER | SYSTEM STARTUP PULSE TIMER
[END INPUTS]
[OUTPUTS]
  SET CONSTANT 10 CNTS
    A272 | Set-Analog (Destination) | ANALOG | ANALOG |
[END OUTPUTS]
```

In this example, the analog A272 is set to a value of 10 when the system is started. This is commonly used to initialize the setpoint value of a DELAY or COUNT equation.


Using an analog setpoint and setting its' value at startup is a good practice to follow. If the panel does not operate properly with the current setpoint value, the analog's value can easily be changed using the front panel keypad.

PRINT a message

This instruction outputs a message to the assigned ports.

Key points to remember:

- Ports include the Panel Display, the Alarm Log, the Trouble Log, LCDs and five printer/CRT ports.
- The first three ports are present in every system.
- Printer/CRT ports are added as needed with the RS232 card.
- It's important to remember that a message output to a **CRT** or to the **Panel Display** is only **MOMENTARILY DISPLAYED**. The CRT will display a message from 2 to 15 seconds. The Panel Display will display a message for about 2 seconds.
- Note: Time displayed may vary depending upon other events occurring in the panel.
- Multiple line print statements are printed as they appear in the programmer, with a carriage return and line feed after each 40 character line.
- If a printer is connected to the first print port the message is printed as follows:
12:03:11 am FRI 01 JAN 90 EMERGENCY EXIT DOOR OPEN



The output of the PRINT equation will read:

```

OUTPUTS:
PRINT <port(s)>:
    <message>
[END-OF-OUTPUTS]

```

Where:

Port(s) is a combination of 1 to 8 print ports selected from a choice list.

Message is a statement typed in the data entry field.

Example:

```

[INPUTS]
    STATUS DETECT
    ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
    PRINT PANEL, TBLLOG, #1 "EMERGENCY EXIT DOOR OPEN"
[END OUTPUTS]

```


In the example, when ZN1 turns ON, the message "EMERGENCY EXIT DOOR OPEN" is momentarily displayed on the Panel LCD and permanently recorded in the Trouble Log with a date & time stamp.

CHLCODE Mapnet channel

This command allows any True Alarm devices "**with piezo bases**" (also referred to as SPHOTO or Sounder Base) on a specific Mapnet or IDnet channel to be coded similar to signal circuits.

Key points to remember:

- This command only sets up the coding rate on the channel it does not actually code the devices. The DEVICE CODE command would be used to select the specific devices that will follow the channel code.
- This command is **used to start the code, as well as stop the code**. So if you use this command to start a code on alarm, you will have to use this command to stop the code on "Signal Silence" or "System Reset".
- **Channel #** is any Mapnet channel (1-10) configured in your system that you wish to code devices on **rate** this is the code that will be played on the channel. The possible choices are:
- The coding options are: OFF, ON, TEMPORAL, MARCH TIME (60 BPM), MARCH TIME (120 BPM), CODING_PATTERN_1 and CODING_PATTERN_2
- The codes for **Coding_Pattern_1** and **Coding_Pattern_2** are for **TFX coding only**.



The output of the equation will read:

```
[OUTPUTS]
  CHLCODE <code> <channel #> PRI==<sp>, <rp>
[END OUTPUTS]
```

Where:

- Output point** is selected from the **Tag List**.
- Code** is a selection made from the **Tag List**.
- Channel #** is a selection made from the **Tag List**
- SP** is the set priority selected from the choice list.
- RP** is the reset priority selected from the choice list.

Example:

```
[INPUTS]
  STATUS ON
  ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
  CHLCODE TEMPORAL M1 PRI=9,9
[END OUTPUTS]
```

To end channel coding custom control will be needed with the OFF code.

DEVICE CODE smart sensors

This command is used to select specific True Alarm devices (with piezo bases) on a Mapnet/IDnet channel that will **follow the channel code** for that channel.

Key points to remember:

- The CHLCODE command would be used to select the specific code that will be played over these devices.
- Due to power consumption, the maximum amount of bases that can be coded per channel is 47.
- Remember to turn off the devices when coding is no longer necessary. Use system lists for off on silence and off on reset.
- All that is required with this command is to select the specific devices and to set the priorities they will be controlled at.

The output of the equation will read:

```
[OUTPUTS]
    DEVCODE PRI==<sp>, <rp>
    <output point>
[END OUTPUTS]
```

Where:

Output point is selected from the **Tag List**.

SP is the set priority selected from the choice list.

RP is the reset priority selected from the choice list.

Example:

```
[INPUTS]
    STATUS ON
    ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
    CHLCODE Coding_Pattern_1 M1 PRI=9,9
    DEVCODE PRI=9,9
    M1-10-0 | SPHOTO | SMOKE | SOUNDER BASE 1
    DEVCODE PRI=9,9
    M1-11-0 | SPHOTO | SMOKE | SOUNDER BASE 2
[END OUTPUTS]
```

It's not necessary for CHLCODE and DEVCODE to be in the same equation, as long as CHLCODE is started somewhere in the Custom Control.

SET ALARM THRESHOLD smart sensors

This command is used to change the Alarm threshold of True Alarm Devices from a specific input, such as either a switch or time control input in custom control.

Key points to remember:

- This can be either a smoke sensor or a heat sensor. When you select either of these choices, a selection of the available settings for the specific type device is displayed.
- Select the setting required, then select the specific Mapnet/IDNet devices you want to change from the tag list.
- **Sensitivity setting** can be either a smoke level for smoke detectors or a temperature setting for heat detectors
- Default sensitivity is 2.5% for photo devices and 135 degrees for heat devices
- Choices for Smoke Sensitivity:

<u>For Mapnet/IDNet Devices</u>				<u>For TFX Devices Only</u>				
0.2	1.0	2.0	3.0	0.6	1.1	2.2	3.1	SMKG1
0.5	1.3	2.5	3.7	0.7	1.2	2.4		SMKG2
0.9	1.5			0.8	1.4	2.6		SMKG3
	1.7				1.6			SMKHI
					1.8			SMKLO
								SMKN
								SMKT2

- Choices for Heat Sensitivity:

<u>For Mapnet/IDNet Devices</u>		<u>For TFX Devices Only</u>
117		200
135		
155		

The output of the equation will read:





```
[OUTPUTS]
    SET THR <qualifier> <%/degree>
        <output point>
[END OUTPUTS]
```

Where:

- Output point** is selected from the **Tag List**.
- Qualifier** is a selection made from the **Tag List**.
- %/Degree** is a selection made from the **Tag List**.

Example:

[INPUTS]
 STATUS ON
 P515 | DIGITAL | UTILITY | SENSITIVITY SWITCH
CONTROL
[END INPUTS]
[OUTPUTS]
 SET THR SMOKE 1.5
 M1-10-0 | SPHOTO | SMOKE | SOUNDER BASE 1
 SET THR SMOKE 1.5
 M1-11-0 | SPHOTO | SMOKE | SOUNDER BASE 2
 SET THR HEAT 117
 M1-12-0 | SHEAT | HEAT | SOUNDER BASE 3
 SET THR HEAT 117
 M1-13-0 | SHEAT | HEAT | SOUNDER BASE 4
[END OUTPUTS]




SET RATE OF RISE heat sensors

This command is used to change the Rate of Rise of True Alarm Heat Detectors from a specific input, such as either a switch or time control input in custom control.

Key points to remember:

- This command is similar to the "Alarm Threshold" command, except that this command is used for only heats, and defines the alarm requirement from a different type parameter.
- When you select this choice, a selection of the available settings is displayed.
- The choices are in **Degrees rise per minute**.
- Select the setting required, then select the specific Mapnet devices you want to change from the tag list.



The output of the equation will read:

```
[OUTPUTS]
    SET RISE <number>
        <output point>
[END OUTPUTS]
```

Where:

Output point is selected from the **Tag List**.
Number is selected from the **Tag List**.

Example:

```
[INPUTS]
    STATUS ON
      P515 | DIGITAL | UTILITY | SENSITIVITY SWITCH CONTROL
[END INPUTS]
[OUTPUTS]
    SET RISE 3
      M1-12-0 | SOHEAT | HEAT | SOUNDER BASE 3
    SET RISE 3
      M1-13-0 | SOHEAT | HEAT | SOUNDER BASE 4
[END OUTPUTS]
```

DISARM 1 or more points

This command is used to ARM and DISARM Priority 2 type points for security applications.

Key points to remember:


- Although the command is called DISARM 1 or more points, depending on the qualifier selected you can ARM or DISARM the selected points
- Selecting **DISARM "ON"** will disarm the selected points.
- Selecting **DISARM "OFF"** will arm the selected points.

The output of the equation will read:

```
[OUTPUTS]
  DISARM <option>
    <output point>
[END OUTPUTS]
```

Where:

Output point is selected from the **Tag List**.
Option is selected from the **Option List**



Example:

Equation 1:

```
[INPUTS]
  STATUS ON
    P520 | DIGITAL | UTILITY |
[END INPUTS]
[OUTPUTS]
  DISARM ON
    ZN7 | SMONB | SPRI2 | MONITOR CARD 4 ZONE ZN7
[END OUTPUTS]
```

Equation 2:

```
[INPUTS]
  NOT STATUS ON
    P520 | DIGITAL | UTILITY |
[END INPUTS]
[OUTPUTS]
  DISARM OFF
    ZN7 | SMONB | SPRI2 | MONITOR CARD 4 ZONE ZN7
[END OUTPUTS]
```

Signal Coding Overview


The signal coding instructions (CODE, MARCH, PNIS and CALIFORNIA, etc) are some of the more complicated output instructions. The best way to understand the operation of the code instruction is to break it down into the individual components.

Group Number

The Code Group Number designates which group the code will be executed in. When a code instruction is executed, the code is put on a queue. As new codes come in, they are stacked up on the queue. Each group has an associated queue.

If only one group is used, the first code instruction to come in will be executed. New codes will be placed on the queue in the order that they come in. When the first code is complete, the next code on the queue will begin execution. This operation (first code in first code out) will continue until the queue is empty. When a single code group is used, only one non-continuous (PNIS) code can be executing at a given time.

Code Group Illustration



If the multiple code groups are used, codes will be queued up in their respective queue. The first code in, first code out operation is the same as with the single code group. The difference is that multiple outputs can be executing different codes at the same time. Multiple code groups might be used in an application where several buildings are to simultaneously play different codes.

The illustration shows how a panel using three code groups would operate. As codes come in they are placed on their respective queue. Code Group 1 could be playing the code 1-1-5 on the signals in list 128 at the same time the code 2-1-2 could be playing on the signals in list 129.

The code groups, numbered 0 thru 7, should not be confused with the lists that are labeled CODING GROUP 0 thru 7. These lists are generated for use in Program 2. Program 2 is a program created by the 4100U Programmer to generate code without the use of User SMPL.

Point List

This is the list of outputs that will play the selected code. The valid output types for coding are SIG, AUX relay, and Master Coding outputs. QAlert NACs are coded differently.

In most cases, **signals and relays should not overlap into different Coding Group output lists.** Overlapping signals may cause garbled code.

Code

There can be up to four digits per code, ranging from 0 to 14. The digits of the code would normally indicate the location of an alarm. The code 2-3-1 might be used to indicate an alarm in Building 2, Floor 3, Area 1.

The code 2-3-1 would turn the signals in the output list on and off twice, at the intervals specified by the **On Time** and **Off Time**. There would then be a pause of one period. A period is equal to the **On Time** plus the **Off Time**. The signals would then turn on and off three times, pause and turn on and off once.

Number of Rounds

The number of rounds is the number of times the code will repeat. There can be from 0 to 15 rounds. Zero rounds means to continually repeat the code.

PNIS (non-continuous) code should not have Number of Rounds equal to zero. New codes would stack up on the queue and not have the opportunity to play, because the present code would continually be repeated.

Number of Pauses

The number of pauses dictates the time between rounds. The following is a list of the selections for Number of Pauses:

0	no pausing
1	1 period
2	3 periods
3	6 periods

NOTE: a period is equal to the On Time plus the Off Time.

On Time, Off Time

The on and off times are self-explanatory. Any combination of the following on/off times can be used:

<u>Seconds</u>	<u>Typical Use</u>
1/4	Fast March Time
1/3	
5/12	California PNIS
1/2	
7/12	Slow March Time
2/3	
3/4	
1.00	
1.25	
1.50	
1.75	
2.00	
2.50	
3.00	
4.00	
5.00	

Set and Reset Priority

The set and reset priorities **DO NOT WORK THE SAME** for the CODE, MARCH, PNIS and CALIFORNIA commands as they do for the HOLD, TRACK and SET commands.

When the input goes true, the output will code at the set priority. If the code gets queued up, it will play at the set priority when it comes off the queue. If the input goes false, the output will continue to code at the set priority. The output will toggle to the reset priority when the code is complete.

Status

The status of a code can be either CONTINUOUS or NON-CONTINUOUS.

A non-continuous code is normally referred to as PNIS (Positive Non-Interfering Successive) code. A non-continuous code can not be interrupted by another code. When a non-continuous code is playing, all new codes are stacked up behind it on the queue. They are stacked up on the queue in the order that they come in. The code command's priority has no bearing on where it sits on the queue.

A continuous code must have the Number of Rounds set to zero. MARCH TIME is an example of a continuous code. If a new code comes in, it will interrupt a continuous code. The continuous code would then be discarded.

If a non-continuous (PNIS) code is playing and a continuous code (MARCH TIME, for example) comes in, the continuous code will wait on the queue for the non-continuous code to complete.

If a continuous code (MARCH TIME, for example) is playing and a non-continuous code (PNIS) comes in, the continuous code will be discarded and the non-continuous code will play.

Tracking Coded Input

Although tracking a coded input is not part of the SMPL equation, coded input does have an effect on SMPL coding.

Coded input comes in from devices that are tied to CODEIN type zones. These devices are mechanical coders which open and close contacts at different intervals. Because they are mechanical, the code they generate can not be queued up like a software code can. Therefore, a CODEIN type zone is allowed to interrupt software generated codes.

The only outputs that can track coded input are the outputs that are in the lists 1 thru 7. All signals/relays that are given the type CODED are auto generated into L1 "CODING GROUP 1 SIGNALS/RELAY". L2 thru L7 can be edited in the Coding Group Configuration Screen of the 4100U Programming Unit. Also in the Coding Group Configuration screen, code groups can be assigned to track code input.

If, for example, a SMPL equation is written to code SIG3 in Group 1 and Group 1 is assigned to track coded input, SIG3 will act one of two ways when a coded input comes in.

Example 1:

```
[INPUTS]
  STATUS DETECT
  ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
  PNIS GROUP=0 ROUNDS=3 1-2-3 PRI=9,9
  L219 | LIST | CONTROL | CODING GROUP 1 SIGNALS/RELAYS
[END OUTPUTS]
```

Assume that SIG3 is in L219. If the above equation came true, SIG3 would begin to put out a 1-2-3 code, but would stop if a coded input came in. If SIG3 was also in L1 "CODING GROUP 1 SIGNALS/RELAYS", it would track the coded input. If SIG3 was not in L1, it would remain silent until the coded input was complete. SIG3 would then put out its 1-2-3 code for 3 rounds.

If the output of the equation in example 1 was a continuous code (i.e. MARCH TIME), SIG3 would march time when the input went true. If a coded input came in and SIG3 was not in L1, SIG3 would continue to march time. If a coded input came in and SIG3 was in L1, SIG3 would track the coded input. When the coded input was complete, SIG3 would continue to march time.

MARCH Code control points

MARCH code is a continuous code, with the on and off times being equivalent.

Key points to remember:

- There are two standard march time codes:
 - Fast March Time = 1/4 sec. on time - 1/4 sec. off time**
 - Slow March Time = 1.5 sec. on time - 1.5 sec. off time**

The output of the equation will read:

```
[OUTPUTS]
  MARCH GROUP=<code group> ON=<on time> OFF=<off time> PRI==<sp>, <rp>
    <list>
[END OUTPUTS]
```

Where:

Code Group is picked from a **choice list** (0 to 7)

On Time is selected from a **choice list**

Off Time is selected from a **choice list**

SP is the set priority selected from the choice list.

RP is the reset priority selected from the choice list.

List is selected from a **Tag List** or entered in the **Data Entry Field**; the list contains the points that are to be coded

Example: (Equation 1 is equivalent to Equation 2)

Equation 1:

```
[INPUTS]
  STATUS DETECT
  ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
  MARCH GROUP=0 ON=1/4 OFF=1/4 PRI=9,9
  L256 | LIST | CONTROL | SPECIAL CODE LIST
[END OUTPUTS]
```

Equation No: 2

```
[INPUTS]
  STATUS DETECT
  ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
  CODE GROUP=0 ROUNDS=0 1 PAUSE=0 ON=1/4 OFF=1/4 STATUS=1 PRI=9,9
  L256 | LIST | CONTROL | SPECIAL CODE LIST
[END OUTPUTS]
```

The MARCH command will automatically set the following parameters:

```
digits = 1
pause = 0
status = CONTINUOUS
```

In the examples above, if ZN1 goes into ALARM, the MARCH code will be put on the queue for Group 0.

If a non-continuous code is being executed in Group 0, the MARCH code will wait for it to finish before coming in.

If there are no codes executing in Group 0 or a continuous code is being executed, the MARCH code will be activated on the signals in L256.

PNIS Code control points

PNIS stands for “Positive Non-Interfering Successive” code. What this means is that a code will be allowed to complete before a new code can come in. This is true for any code operation whose status is NON-CONTINUOUS.

Key points to remember:

- **CODED INPUT WILL INTERRUPT** a PNIS code. If a code group is assigned to track coded input, by the programmer, and a CODEDIN type zone comes in while a PNIS code is playing, the PNIS code will be interrupted. When the coded input is complete, the interrupted code will be reactivated and will code all of the specified number of rounds.

The output of the equation will read:

```
[OUTPUTS]
    PNIS GROUP=<code group> ROUNDS=<rounds> <digits> PRI=<sp>, <rp>;
    <list>
[END OUTPUTS]
```

Where:

Code groups are numbered from 0 to 7.

Rounds are the number of times that the code will repeat. The number of rounds can be from 0 to 15.

Digits are the actual code and can be from 0 to 14. There can be up to four digits per code.

SP is the set priority and can be from 2 to 15.

RP is the reset priority and can be from 2 to 15.

List is selected from the **Tag List** or entered in the data entry field. This list contains the points that are to be coded.

Example: (Equation 1 is equivalent to Equation 2)

Equation 1:

```
[INPUTS]
    STATUS DETECT
    ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
    PNIS GROUP=1 ROUNDS=3 1-2-3 PRI=6,9
    L256 | LIST | CONTROL | SPECIAL CODE LIST
[END OUTPUTS]
```

Equation 2:

```
[INPUTS]
    STATUS DETECT
    ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
    CODE GROUP=1 ROUNDS=3 1-2-3 PAUSE=2 ON=1/2 OFF=1/2 STATUS=0 PRI=6,9
    L256 | LIST | CONTROL | SPECIAL CODE LIST
[END OUTPUTS]
```

The two equations above would achieve the same results. The PNIS command will automatically set the following parameters:

pause = 2
on time = 1/2
off time = 1/2
status = NON-CONTINUOUS

In the examples above, if ZN1 goes into ALARM the PNIS code will be put on the queue for Group 1.

If a non-continuous code is being executed in Group 1, the PNIS code will wait for it to finish before coming in.

If there are no codes executing in Group 1 or a continuous code is being executed, the PNIS code will be activated on the signals in L256.

CALIFORNIA Code control points

CALIFORNIA code is a special coding operation requested by the California region.

Key points to remember:

- There are two standard California codes:
 - 1 minute code- rounds = 4**
 - 3 minute code- rounds = 12**

The output of the equation will read:

```
[OUTPUTS]
    CALIF GROUP=<code group> ROUNDS=<rounds> PRI=<sp>, <rp>
    <list>
[END OUTPUTS]
```

Where:

Code groups are numbered from 0 to 7.

Rounds are the number of times that the code will repeat. The number of rounds can be from 0 to 15.

SP is the set priority and can be from 2 to 15.

RP is the reset priority and can be from 2 to 15.

List is selected from the **Tag List** or entered in the data entry field. The list contains the points that are to be coded.

Example: (Equation 1 is equivalent to Equation 2)

Equation 1:

```
[INPUTS]
    STATUS DETECT
    ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
    CALIF GROUP=1 ROUNDS=4 PRI=9,9
    L256 | LIST | CONTROL | SPECIAL CODE LIST
[END OUTPUTS]
```

Equation 2:

```
[INPUTS]
    STATUS DETECT
    ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
    CODE GROUP=1 ROUNDS=4 1-2 PAUSE=3 ON=5/12 OFF=5/12 STATUS=0 PRI=9,9
    L256 | LIST | CONTROL | SPECIAL CODE LIST
[END OUTPUTS]
```

The two equations above would achieve the same results. The CALIFORNIA command will automatically set the following parameters:

```
digits = 12
pause = 3
on time = 5/12
off time = 5/12
status = NON-CONTINUOUS
```

In the examples above, if ZN1 goes into ALARM the CALIFORNIA code will be put on the queue for Group 1.

If a non-continuous code is being executed in Group 1, the CALIFORNIA code will wait for it to finish before coming in.

If there are no codes executing in Group 1 or a continuous code is being executed, the CALIFORNIA code will be activated on the signals in L256.

CODE (generic) control points

The generic CODE instruction is the all-purpose coding operator.

Key points to remember:

- The other three code instructions (MARCH, PNIS and CALIFORNIA) are all derived from the generic code operator, with several of their parameters automatically set. In other words, MARCH, PNIS and CALIFORNIA code are all short cuts to writing a CODE equation with all of the required parameters.

The output of the equation will read:

```
[OUTPUTS]
  CODE GROUP=<code group> ROUNDS=<rounds> PAUSE=<pause> ON=<time on>
  OFF=<time off> STATUS= <status> PRI==<sp>, <rp>
  <list>
[END OUTPUTS]
```

Where:

Code groups are numbered from 0 to 7.

Rounds are the number of times that the code will repeat. The number of rounds can be from 0 to 15.

Digits are the actual code and can be from 0 to 14. There can be up to four digits per code.

Pause is the number of wait states between rounds. A wait state is equal to the <time on> plus the <time off>. There can be from 0 to 3 wait states.

On time and **Off time** are selected from the choice list in the data entry field.

Status can be continuous (1) or non-continuous (0).

SP is the set priority and can be from 2 to 15.

RP is the reset priority and can be from 2 to 15.

List is selected from the **Tag List** or entered in the data entry field. The list contains the points that are to be coded.

Example: (Temporal Code)

Equation 1:

[INPUTS]

STATUS DETECT

ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1

[END INPUTS]

[OUTPUTS]

CODE GROUP=0 ROUNDS=0 3 PAUSE=2 ON=1/2 OFF=1/2 STATUS=1 PRI=9,9

L256 | LIST | CONTROL | SPECIAL CODE LIST

[END OUTPUTS]

In this example the generic code operator is used to put out a Temporal Code.

The following parameters must be set for Temporal Code:

Group Number: 0	Off Time (secs): 1/2
Number of Rounds: 0	Status: CONTINUOUS
Code: 3	Set Priority: 9
Number of Pauses: 2	Reset Priority: 9
On Time (secs): 1/2	Point List:

UPLAY or UPLAYQ audio messages

Key points to remember:

- UPLAY
 - o UPLAY immediately plays a message on the specified channel clearing all messages which are queued up to be played.
 - o The message will only be played if the priority specified is high enough to take precedence over any message that is playing.
 - o The list of valid channels is based on the number of channels configured for the system.
 - o All of the channels except for the optional second channel in an analog system can play any of the available messages.
 - o The optional second channel in an analog audio system is limited in what it can play.


- UPLAYQ
 - o UPLAYQ queues up a message to be played on the specified channel after all previously queued have been played.
 - o The priority specified is used to determine the importance of the message.
 - o Valid channels and messages are the same for the UPLAYQ as they are for the UPLAY.

The best way to understand the UPLAY command is to break it down into its' individual components.

Channel


There can be up to eight channels per job. An audio card configured for single channel operation has one channel, which is referred to as CHL1.

Selecting a specific channel will determine which channel the specified message will come from.



Attribute


The attribute setting is similar to the route command in that it can be used to specify the audio source to play.



Message Number

The message number indicates what message is to be played. It can be a numeric value or an analog pseudo which holds the value of the message number. The message numbers are assigned in the audio message and can be changed in the message editor.

The message can be set using a message number defined in the chip set or by using a system analog pseudo to play a preset message.



The message assigned to each of analog in assigned with the message editor.

System analogs that can be used are:

- A54 = Supervision MSG#
- A55 = Evacuation MSG#
- A56 = Alert MSG#
- A57 = Drill MSG#
- A58 = All Clear MSG#
- A59 = Aux1 MSG#
- A60 = Aux2 MSG#
- A61 = Microphone Pretone MSG#

UVOLUME

This changes the volume of the amplifier controlling the selected speaker(s). The options available for volume control are LOW, MEDIUM, HIGH, and RAMP.

When using A208, the UVOLUME analog pseudo the volume is set based on the value contained in the analog pseudo. The values for the analog pseudo are:

- 0 – No Effect (not used in UVOLUME)
- 1 – Low
- 2 – Medium
- 3 – High
- 4 – Ramp

Reset Speaker Circuits?

Selecting this option will cause the speakers in the Speaker List to turn off when the message is complete. It is equivalent to using the USPEAKER OFF command.

Flush (remove) queued audio commands?




Selecting this option will cause all queued messages (PNIS or PLAYQ) to be flushed from the play queue.

Select Speaker List (Optional)

The Speaker List choice is optional. The selected speakers will turn on for the duration of the message playing.

Tracking Pseudo

The Tracking Pseudo choice is optional. The selected digital pseudo will be on for the duration of the message playing. The Tracking Pseudo is always at a priority zero.



Set Priority**Reset Priority**

The set and reset priorities **DO NOT work the same for the PLAY equations** as they do for the HOLD, TRACK and SET commands. The channel routing, and signal output point priority levels will be at the set priority for the duration of the message playing. When the message is complete, the priority levels will toggle to the reset priority.

**Example:**

```
[INPUTS]
  STATUS ON
    P520 | DIGITAL | UTILITY |
[END INPUTS]
[OUTPUTS]
  UPLAY CHL1=1 VOLUME=NOEFFECT NORESET NOFLUSH SPK=L258 TRK=P600 PRI=7,9
[END OUTPUTS]
```

CHL1=1

If P520 turns on, message 1 will play immediately on Ch11 if the PRIORITY is greater than or equal to 7.

Volume=NOEFFECT

The volume will not change from the last message that was played.

NOFLUSH

Queued messages won't be cleared to play this message.

NORESET

Speakers will not turn off when the message is complete.

L258

The speaker circuits in list 258 will be routed to the CHL1 input if the priority level of the output point is greater than or equal to 7.

trk=P600

The tracking pseudo P600 will turn on at a priority 0 when the message starts. It will turn off at a priority 0 when the message is complete.

pri=7,9

The priority levels will toggle to a 9 when the message is complete.

UROUTE audio

This is used to route a channel to an audio source

Key points to remember:

- The set and reset priorities DO NOT work the same for the UROUTE command as they do for the HOLD, TRACK, and SET commands.
- Normally, depending on the application, the set priority is higher than the output channel's current priority in order for the command to take effect.
- The output channel's priority level will immediately toggle to the reset priority. The condition of the input has no bearing on the reset priority.
- To get the control of the audio back to the panel's audio controller, you must route the output channel to silence (i.e. UROUTE CHL1=SILENCE PRI=9,9)

The output of the equation will read:

```
[OUTPUTS]
    UROUTE <output channel>=<input port> PRI==<sp>, <rp>
[END OUTPUTS]
```

Where:

Output Channel can be CHL1 – CHL8

Input Port can be Silence, Phone, Mike, Rmike, or any 1 of 8 Option Card inputs.

SP is the set priority and can be from 2 to 15.

RP is the reset priority and can be from 2 to 15.

Example:

```
[INPUTS]
    STATUS ON
    P520 | DIGITAL | UTILITY |
[END INPUTS]
[OUTPUTS]
    UROUTE CHL2=OPCARD1_1 PRI=8,9
[END OUTPUTS]
```

In this example, when P520 turns on, Option Card 1, input 1 will be routed to channel 2 if the CHL2 priority level allows. Immediately after routing occurs the CHANNEL 2 ROUTING PRIORITY will toggle to the reset priority of 9.

USILENCE audio channel

This command will turn off the selected channel by playing silence to a specific channel.

Key points to remember:

- This command does not flush the queue. Any UPLAYQ or UPNIS commands waiting on the queue will be played if the SILENCE command is issued.
- The set and reset priorities DO NOT work the same for the SILENCE command as they do for the HOLD, TRACK, and SET commands.
- Normally, depending on the application, the set priority is higher than the output channel's current priority in order for the command to take effect.
- The channel priority level will immediately toggle to the reset priority. The condition of the input has no bearing on the reset priority.

The output of the equation will read:

```
[OUTPUTS]
    USILENCE <chl> PRI==<sp>, <rp>
[END OUTPUTS]
```

Where:

Chl can be channels 1 through 8

SP is the set priority and can be from 2 to 15.

RP is the reset priority and can be from 2 to 15.

Example:

```
[INPUTS]
    STATUS ON
    P6 | DIGITAL | UTILITY | ALARM SILENCE
[END INPUTS]
[OUTPUTS]
    USILENCE CHL1 PRI=8,9
[END OUTPUTS]
```

P6 turns on when the signal silence key is hit. When P6 turns on, The channel 1 is silenced, the channel priority level will toggle to the reset priority of 9. Any messages that are queued up on channel will then be play, if the priority allows.

USPEAKER circuits on/off

The USPEAKER command is very similar to the HOLD command. The difference being that the USPEAKER command provides for the routing of the output to one of eight channels.

Key points to remember:

- USPEAKER turns on the specified speaker circuits and routes them to the specified channel (including Audio Riser 2).
- The set and reset priorities DO NOT work the same for the SPEAKER command as they do for the HOLD command.
- Normally, depending on the application, the set priority is higher than the output channel's current priority in order for the command to take effect. This is true for all prioritized commands.
- The output point's priority level will **immediately toggle to the reset priority**. The status of the input has no bearing on the reset priority.

The output of the equation will read:

```
[OUTPUTS]
    USPEAKER <input channel> PRI==<sp>, <rp>
        <list>
[END OUTPUTS]
```

Where:

Input Channel can be CHL1 – CHL8, ON or OFF.

Output Point is selected from the Tag List or entered in the data entry field.

SP is the set priority and can be from 2 to 15.

RP is the reset priority and can be from 2 to 15.

Example:

```
[INPUTS]
    STATUS ON
    P520 | DIGITAL | UTILITY |
[END INPUTS]
[OUTPUTS]
    USPEAKER CHL2 PRI=8,9
    L258 | LIST | CONTROL | SPECIAL SPEAKER LIST
[END OUTPUTS]
```

In this example, when P520 turns on the speaker circuits, L258 will be turned on to channel 2 if the priority level allows. Immediately after the signal turns on, the priority levels will toggle to the reset priority of 9.

UPNISV voice code speaker circuits

This command is used to generate spoken codes. This specific PNIS opcode (PNISV – the “V” stands for voice) causes the playing of the “spoken” numbers for each of up to 4 digits specified.

Key points to remember:

- Voice coding requires chipset 3, or a custom chip set. Coding is **not** available in chipset 42.
- All PNISx opcodes can optionally play a preamble or introduction message, immediately followed by the specified tone or voice output, immediately followed by a postamble message.

The best way to understand the UPNISV command is to break it down into its' individual components.

Preamble Message Number

The Preamble can be any valid message that has this option selected in the message editor. It can be a numeric or analog value. The preamble should not be a "forever" message; this would cause the spoken code not to be heard until the channel was silenced.

The preamble would normally be a message which leads into the spoken code, such as:

"DR. RED, YOUR ASSISTANCE IS
REQUIRED IN AREA"

Code

The Code is simply the spoken digits which follow the preamble. There can be up to four digits.

Postamble Message Number





The Postamble can be any valid message that has this option selected in the message editor. It can be a numeric or analog value. The postamble should not be a "forever" message, this would cause only the first round of the PNISV message to play. Any subsequent PNISV message would not be heard until the channel was silenced.

The postamble follows the spoken code. A common postamble would be a message such as:

"RESPOND IMMEDIATELY"

Number of Rounds

The Number of Rounds is the number of times that the entire message is repeated. Number of Rounds can be from 0 to 15, where 0 means repeat until silenced.



Set Priority**Reset Priority**

The set and reset priorities DO NOT work the same for the PNISV equations as they do for the HOLD, TRACK and SET commands. The channel routing, and signal output point priority levels will be at the set priority for the duration of the message playing. When the message is complete, the priority levels will toggle to the reset priority.

Example:

```
[INPUTS]
  STATUS ON
    P520 | DIGITAL | UTILITY |
[END INPUTS]
[OUTPUTS]
  UPNISV CHL1=(MSG=100 DIGITS=1 MSG=98) ROUNDS=2 PRI=7,9
    L258 | LIST | CONTROL | SPECIAL SPEAKER LIST
[END OUTPUTS]
```

M100 = "ATTENTION, AN EMERGENCY HAS BEEN REPORTED ON FLOOR"

M98= "RESPOND IMMEDIATELY"

When P520 turns on the PNISV equation is put on the channel 1 queue.

When the PNISV equation comes off the queue:

CHL1

If the current value of channel 1 is greater than or equal to 7, the channel 1 output of the Audio Card is at a priority 7.

msg=(M100 M98)

If the current value of channel 1 is greater than or equal to 7, the message, "ATTENTION, AN EMERGENCY HAS BEEN REPORTED ON FLOOR ONE.RESPOND IMMEDIATELY." will play on channel 1 at a priority 7.

rounds=2

The message will play for two rounds.

pri=7,9

When the message is complete, the CHANNEL 1 and the speaker output priority levels will drop to a 9.

L258

If the current value of the priority level of the output points in list 258 is greater than or equal to 7, the speaker circuits in list 258 are routed to their channel 1 input at a priority 7.

UPNISM message code speaker circuits

This command is used to string, up to three, message together. This specific PNIS opcode (PNISM – the “M” stands for message) causes the playing of the “spoken” messages specified.

Key points to remember:

- Up to three messages can be specified.
- The messages will be played one after another in the order specified.
- These messages are sandwiched between the optional preamble and postamble messages.

This code is similar in structure to the PNISV and PNIST codes.

Preamble Message Number

The Preamble can be any valid message that has this option selected in the message editor. It can be a numeric or analog value. **The preamble should not be a "forever" message**, this would cause the Custom Message not to be heard until the channel was silenced.

The preamble would normally be a message which leads into the Custom Message, such as:
"ATTENTION, AN EMERGENCY HAS BEEN REPORTED AT THE"

Custom Message

The Custom Message can be any valid message that has this option selected in the message editor. It can be a numeric value or analog value. The Custom Message is the second message in a string of three messages.

The Custom Message would normally be a message created specifically for a particular job. The Custom Message may be a location on the site where the panel will be installed, such as:
"HANCOCK BUILDING"

Postamble Message Number

The Postamble can be any valid message that has this option selected in the message editor. It can be a numeric or analog value. **The postamble should not be a "forever" message**; this would cause only the first round of the PNISM message to play. Any subsequent PNISM message would not be heard until the channel was silenced.

The postamble follows the Custom Message. A common postamble would be a message such as:
"RESPOND IMMEDIATELY"

If no postamble is required, enter message number zero.

Example:

```
[INPUTS]
  STATUS ON
    P520 | DIGITAL | UTILITY |
[END INPUTS]
[OUTPUTS]
  UPNISM CHL1=(MSG=100 MSG=110 MSG=98) ROUNDS=2 PRI=7,9
    L258 | LIST | CONTROL | SPECIAL SPEAKER LIST
[END OUTPUTS]
```

M100 = "ATTENTION, AN EMERGENCY HAS BEEN REPORTED AT THE"

M110 = "HANCOCK BUILDING"

M98 = "RESPOND IMMEDIATELY"

When P520 turns on the PNISM equation is put on the channel 1 queue.

When the PNISM equation comes off the queue:

CHL1

If the current value of channel 1 is greater than or equal to 7, the channel 1 output of the Audio Card is at a priority 7.

msg=(M100 M110 M98)

If the current value of channel 1 is greater than or equal to 7, the message, "ATTENTION, AN EMERGENCY HAS BEEN REPORTED AT THE HANCOCK BUILDING. RESPOND IMMEDIATELY." will play on channel 1 at a priority 7.

Rounds

The message will play for two rounds.

pri=7,9

When the message is complete, the CHANNEL 1 and the speaker output priority levels will drop to a 9.

L258

If the current value of the priority level of the output points in list 258 is greater than or equal to 7, the speaker circuits in list 258 are routed to their channel 1 input at a priority 7.

QCODE

QCODE is used with QALERT device types, TrueAlert Non-Addressable points.

Key points to remember:


- The STROBESON opcode gives the ability to controls horns and strobes separately on the same two wire device.
- The QALERT device would be held on with custom control on a user specified event. To silence the horn but allow the strobes to continue the flash the QCODE opcode with STROBEON qualifier would be used.

The output of the equation will read:

```
[OUTPUTS]
    QCODE STROBESON PRI==<sp>, <rp>
        <list>
[END OUTPUTS]
```

Where:

Code groups are numbered from 0 to 7.



Example:

```
[INPUTS]
    STATUS ON
      P6 | DIGITAL | UTILITY | ALARM SILENCE
    AND STATUS ON
      SIG4 | SIGB | QALERT | SIGNAL CARD 1 CIRCUIT SIG4
[END INPUTS]
[OUTPUTS]
    QCODE STROBESON PRI=9,9
      SIG4 | SIGB | QALERT | SIGNAL CARD 1 CIRCUIT SIG4
[END OUTPUTS]
```



This equation stops the horn but keeps the strobes going.

UVOLUME


This changes the volume of the amplifier controlling the selected speaker(s).

Key points to remember:

- The options available for volume control are LOW, MEDIUM, HIGH, and RAMP.
- When using A208, the UVOLUME analog pseudo the volume is set based on the value contained in the analog pseudo. The values for the analog pseudo are:
 - 0 – No Effect (not used in UVOLUME)
 - 1 – Low
 - 2 – Medium
 - 3 – High
 - 4 – Ramp

The output of the equation will read:

```
[OUTPUTS]
  UVOLUME VOLUME=<option>
    <list>
[END OUTPUTS]
```



Where:

Option is a selection from the **Tag List**.

Example:

```
[INPUTS]
  STATUS FIRE
  L18 | LIST | EMPTY | GENERAL FIRE ALARM MONITOR ZONES
[END INPUTS]
[OUTPUTS]
  UVOLUME VOLUME=HIGH
  L90 | LIST | EMPTY | ALL SPEAKERS - NON EDITABLE
[END OUTPUTS]
```

UPNIST tone code speaker circuits

This command is used to generate tone codes (HORN, BELL or CHIME). This specific PNIS opcode (PNIST – the “T” stands for tone) causes the playing of a number beats of a tone which can be counted out to determine the up to 4 digits specified. This is sandwiched between the optional preamble and postamble codes.

This code is similar in structure to the PNISV code.

Tone

The choices of tones are:

M1 = HORN

M2 = BELL

M3 = CHIME

Code

There can be up to four digits per code, ranging from 0 to 14. The digits of the code would normally indicate the location of an alarm.

The code 1-2-3 might be used to indicate an alarm in Building 1, Floor 2, Area 3.

Example:

```
[INPUTS]
  STATUS ON
    P520 | DIGITAL | UTILITY |
[END INPUTS]
[OUTPUTS]
  UPNIST CHL1=(TONE=2 DIGITS=1-2-3) ROUNDS=4 PRI=7,9
    L258 | LIST | CONTROL | SPECIAL SPEAKER LIST
[END OUTPUTS]
```

When P520 turns on the PNIST equation is put on the channel 1 queue.

When the PNIST equation comes off the queue:

CHL1

If the current value of channel 1 is greater than or equal to 7, the channel 1 output of the Audio Card is at a priority 7.

msg=BELL (1-2-3)

If the current value of the channel is greater than or equal to 7, the code 1-2-3 will play on channel 1 at a priority 7. The tone of the 1-2-3 code will be the BELL.

Rounds



The message will play for four rounds.

pri=7,9

When the message is complete, the CHANNEL 1 and the speaker output priority levels will drop to a 9.

L258

If the current value of the priority level of the output points in list 258 is greater than or equal to 7, the speaker circuits in list 258 are routed to their channel 1 input at a priority 7.



MARCH60

MARCH60 is a special OPCODE designed to code QLAERT and True Alert virtual NACs.

Key points to remember:

- When using this OPCODE points to be coded should be placed in a list. If a custom control is written against an individual point and a list is required an invalid OPCODE usage error will occur.

The output of the equation will read:

```
[OUTPUTS]
  MARCH60 PRI==<sp>, <rp>
    <list>
[END OUTPUTS]
```

Where:


SP is the set priority and can be from 2 to 15.

RP is the reset priority and can be from 2 to 15.

List is selected from the **Tag List** or entered in the data entry field.

Example:

```
[INPUTS]
  STATUS DETECT
  ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
  MARCH60 PRI=9,9
  L262 | LIST | CONTROL | 4009A4
[END OUTPUTS]
```



MARCH120

MARCH120 is a special OPCODE designed to code True Alert virtual NACs.

Key points to remember:

- If QALERT devices are programmed using the MARCH120 OPCODE they will only play a MARCH60 code.
- When using this OPCODE points to be coded should be placed in a list. If a custom control is written against an individual point and a list is required an invalid OPCODE usage error will occur.

The output of the equation will read:

```
[OUTPUTS]
  MARCH120 PRI==<sp>, <rp>
    <list>
[END OUTPUTS]
```

Where:


SP is the set priority and can be from 2 to 15.

RP is the reset priority and can be from 2 to 15.

List is selected from the **Tag List** or entered in the data entry field.

Example:

```
[INPUTS]
  STATUS DETECT
  ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
  MARCH120 PRI=9,9
  L259 | LIST | CONTROL | 4009T DEVICES
[END OUTPUTS]
```



TEMPORAL

TEMPORAL is a special OPCODE designed to code QALERT and True Alert virtual NACs.

Key points to remember:

- When using this OPCODE points to be coded should be placed in a list. If a custom control is written against an individual point and a list is required an invalid OPCODE usage error will occur.

The output of the equation will read:

```
[OUTPUTS]
    TEMPORAL PRI==<sp>, <rp>
        <list>
[END OUTPUTS]
```

Where:


SP is the set priority and can be from 2 to 15.

RP is the reset priority and can be from 2 to 15.

List is selected from the **Tag List** or entered in the data entry field.

Example:

```
[INPUTS]
    STATUS DETECT
    ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
    TEMPORAL PRI=9,9
    L260 | LIST | CONTROL | 4009T
[END OUTPUTS]
```



MARCH20

MARCH20 is a special OPCODE designed to code QALERT and True Alert virtual NACs.

Key points to remember:

- When using this OPCODE points to be coded should be placed in a list. If a custom control is written against an individual point and a list is required an invalid OPCODE usage error will occur.

The output of the equation will read:

```
[OUTPUTS]
MARCH20 PRI==<sp>, <rp>
      <list>
[END OUTPUTS]
```

Where:


SP is the set priority and can be from 2 to 15.

RP is the reset priority and can be from 2 to 15.

List is selected from the **Tag List** or entered in the data entry field.

Example:

```
[INPUTS]
STATUS DETECT
  ZN1 | SMONB | FIRE | MONITOR CARD 4 ZONE ZN1
[END INPUTS]
[OUTPUTS]
MARCH20 PRI=9,9
  L261 | LIST | CONTROL | TA SUPPLY
[END OUTPUTS]
```



HPULSE

The HPulse opcode is used to pulse a control point ON for a full duration of 1 to 15 seconds regardless of the state of input.

Key points to remember:

- HPulse is an opcode that is used on 2120 to **4100U BMUX conversions**.
- HPulse can be done using a defined constant, or an analog can be used which would be set from another equation and can be changed from the front panel.

The output of the equation will read:

```
[OUTPUTS]
  HPULSE CONSTANT <seconds> PRI==<sp>, <rp>
      <bmux point>
      <analog point>
[END OUTPUTS]
```

Where:

Seconds is entered in the **data entry field**.

SP is the set priority and can be from 2 to 15.

RP is the reset priority and can be from 2 to 15.




Bmux point is selected from the **Tag List**

Analog point is selected by the programmer (next available point)

Example:

```
[INPUTS]
  STATUS ON
  101-101 | INPUT | USWITCH |
[END INPUTS]
[OUTPUTS]
  HPULSE CONSTANT 15 PRI=9,9
  101-201 | BMUX Ctrl Point. | PHOTO
  A272 | Analog Counter | ANALOG | ANALOG |
[END OUTPUTS]
```

When 101-101 is active, HPulse will turn on output 101-201 for time specified of 15 seconds. The output pulses for specified time even if input is active for a shorter duration than the pulse time specified.



TCODE on a list of TrueAlert points

TCODE is designed to do on the fly changes to the type of coding being done on TrueAlert VNACs. When executed on an activated VNAC the default coding specified changes to the code specified by the TCODE operation.

Key points to remember:

- TCODE works only on activated VNACs and once the VNAC is deactivated the coding reverts back to the programmed default.
- Once active, the TCODE code stays in its state until either another TCODE command is executed or until the VNAC is silenced or reset. Setting TCODE to OFF will cause the coding to revert back to its original type.
- TCODE is intended for operations where horn cadence needs to be changed from one pace to another.
- Remember that the VNAC devices must be placed into a list for use in custom control when using this opcode.
- On Revision 12.04 systems and higher, it is possible to change the coding type on the fly using the TCODE SMPL op code. When executed on an activated VNAC, this op code changes the default coding specified in this section to a different code specified by the TCODE command. TCODE only works on activated VNACs, and once the VNAC is deactivated the coding reverts back to the programmed default. TCODE is intended for operations like recall, where the cadence of the horns is changed when the incident is over to recall people into the building. It should be noted that when a TCODE command is executed the VNAC stays in that state until another TCODE command is executed, or until the VNAC is deactivated. Thus, if override of the TCODE induced cadence is required on activation of a subsequent alarm that must be programmed using another TCODE command. Setting TCODE to OFF will cause the coding type to revert back to the original coding type.

The output of the equation will read:

```
[OUTPUTS]
    TCODE <list> <code> PRI==<sp>, <rp>
[END OUTPUTS]
```


Where:

List is selected from a **Tag List**

Code is selected from a **Choice List**

SP is the set priority and can be from 2 to 15.

RP is the reset priority and can be from 2 to 15.



Example:

```
[INPUTS]
    STATUS ON
    P560 | DIGITAL | UTILITY |
[END INPUTS]
[OUTPUTS]
    TCODE L257 TEMPORAL PRI=9,9
[END OUTPUTS]
```

ACKNOWLEDGE TBL/SUP/FIRE/PRI2

ACKN is designed to Acknowledge Fire, Supervisory, Trouble or Priority 2 events from something other than an acknowledge key.

Key points to remember:

- This OPCODE is used to acknowledge events from some other source than the standard acknowledge keys.
- This would be used on a 4100U setup as a Slave Panel in an XA Loop configuration with a XALIC card. Acknowledge would be received from the XA panel and turn on a digital pseudo in the 4100U.

The output of the equation will read:

```
[OUTPUTS]  
    ACKN <event>  
[END OUTPUTS]
```

Where:

Event is a choice of either **FIRE** or **SUPRV** or **TROUBLE** or **PRI2ALM**

Example:


```
[INPUTS]  
    STATUS ON  
    P512 | DIGITAL | UTILITY | FIRE ACKNOWLEDGE FROM AL-1500  
[END INPUTS]  
[OUTPUTS]  
    ACKN FIRE  
[END OUTPUTS]
```

Chapter 3: Additional Coding

It should also be noted that doing Audio or DC coding is not limited to writing custom control.


Key points to remember:

- PNIS codes can be directly applied to an alarm point in the PNIS Code of the point configuration screen. This can be either a numeric code sequence of 4 digits between 0 and 14, or a message that has been identified for coding in the chip set.
- Once PNIS codes have been assigned to the alarm points coding properties can be configured in the Coding sub-tab on the List screen.




Audio Coding

The properties screen can be accessed by selecting the **Audio coding group** and either double clicking on or right clicking and selecting properties.



Coding can be selected for **Monitor** and **Control** groups to **auto generate** to include all points or select as required.



From the **Audio Group** tab there is a choice of doing **Tone** or **Voice coding**.


- When selecting **tones coding**; **pre** and **post codes** can be selected. Typically the tone would be used an alert to identify the beginning and end of the coded message.

Operation mode can be changed from **On Til Completion, Reset, and Silence.**

Rounds from **0 to 15** can be selected with zero being equal to continuous operation.

The **Tone** selected would be used to sound the numeric sequence entered. Available tones are **Bell, Horn** and **Chime.**


A 1-3-1 code with bells selected will sound a bell once, pause, bell 3 times, pause, bell 1 time.



- When selecting **audio (voice) coding**; **pre** and **post codes** can be selected as either tones or voice. Typically tones to be used an alert to identify the beginning and end of the coded message where voice would be used to string phrases together.

Operation mode can be changed from **On Til Completion, Reset, and Silence.**

Rounds from **0 to 15** can be selected with zero being equal to continuous operation.



DC Coding



DC Coding (direct coding) occurs when a coded input device is used and coded output signals are selected.

When these devices are in use DC coding can be configured in the Coding sub-tab on the List screen. The properties screen can be accessed by selecting the DC coding group and either double clicking on or right clicking and selecting properties.

DC coding can use used to follow a coded input or code using specific parameters including On Til, number of rounds, On Time and Off Time.

On Time; Allows you to set the duration of each tone within the PNIS code.

Off Time; Allows you to set the duration of silence between tones of the PNIS code.



Chapter 4: Example Programming

AHU Staggered Start

The following Example represents an AHU STAGGERED Start program. This is a good example of how the cycle command is used. For this application you would need to:

Edit List:

L89 “AHU RELAYS OFF ON RESET “- REMOVE all AHU relays from list L89

Create Digital Pseudos :

P512 | DIGITAL AHU START- UP IN PROGRESS (utility point type)

P513 | DIGITAL AHU START-UP ENABLE (utility point type)

Custom Control Notes:

1. Analog Pseudo A256 is set to a value of zero.
2. Analog Pseudo A257 is set to the delay time desired, between relays. This example has a delay time of 10 seconds.
3. Analog pseudo A258 is the running timer.
4. Analog Pseudo A259 is the counter whose increments turn off a relay. It should be set to a value of 100 unless the amount of AHU relays in the system exceeds 100.

Example:

Equation 1: System Start-up

[INPUTS]

STATUS ON

A34 | ANALOG | TIMER | SYSTEM STARTUP PULSE TIMER

[END INPUTS]

[OUTPUTS]

SET CONSTANT 0 SECS

A256 | Set-Analog (Destination) | ANALOG | ANALOG | IS SET TO A VALUE OF 0

SET CONSTANT 10 SECS

A257 | Set-Analog (Destination) | ANALOG | ANALOG | IS SET TO THE DELAY TIME DESIRED

[END OUTPUTS]

Equation 2: Start-up Inhibit

[INPUTS]

STATUS ON

A0 | ANALOG | ANALOG | NUMBER OF SYSTEM FIRE ALARMS

[END INPUTS]

[OUTPUTS]

HOLD ON PRI=9,9

P513 | DIGITAL | UTILITY | AHU START-UP ENABLE

[END OUTPUTS]

Equation 3: System Reset

[INPUTS]

STATUS ON

A21 | ANALOG | TIMER | SYSTEM RESET PULSE TIMER

AND STATUS ON

P513 | DIGITAL | UTILITY | AHU START-UP ENABLE

[END INPUTS]

[OUTPUTS]

HOLD ON PRI=9,9

P512 | DIGITAL | UTILITY | AHU START- UP IN PROGRESS

[END OUTPUTS]

Equation 4: Cycle Command

```

[INPUTS]
  STATUS ON
    P512 | DIGITAL | UTILITY | AHU START- UP IN PROGRESS
  CYCLE ANALOG
    A258 | Analog Timer | ANALOG | ANALOG | IS THE RUNNING TIMER
    A256 | Analog ON Timer SetPoint | ANALOG | ANALOG | IS SET TO A VALUE OF 0
    A257 | Analog OFF Timer SetPoint | ANALOG | ANALOG | IS SET TO THE DELAY TIME DESIRED
[END INPUTS]
[OUTPUTS]
  PRINT ALL "AHU START-UP IN PROGRESS"
  TRACK ON PRI=9,9
    0-7-2 | MCODE7 | PIEZO | Master Output - Display Sonalert
[END OUTPUTS]

```

Equation 5: Count

```

[INPUTS]
  STATUS ON
    A258 | ANALOG | ANALOG | IS THE RUNNING TIMER
  COUNT CONSTANT 100 RESET ON
    A259 | Analog Counter | ANALOG | ANALOG | COUNTER WHOSE INCREMENTS TURN RELAY OFF
    P3 | Reset-Point | DIGITAL | UTILITY | FIRE ALARM DETECT
[END INPUTS]
[OUTPUTS]
[END OUTPUTS]

```

Equation 6: 1st Reset

```

[INPUTS]
  IF ANALOG = CONSTANT 1.000000 CNTS
    A259 | Compare Analog | ANALOG | ANALOG | COUNTER WHOSE INCREMENTS TURN RELAY OFF
[END INPUTS]
[OUTPUTS]
  HOLD OFF PRI=9,9
    AUX5 | RELAY | AHUR | AIR HANDLING UNIT RELAY 1
[END OUTPUTS]

```

Equation 7: 2nd Reset

```

[INPUTS]
  IF ANALOG = CONSTANT 2.000000 CNTS
    A259 | Compare Analog | ANALOG | ANALOG | COUNTER WHOSE INCREMENTS TURN RELAY OFF
[END INPUTS]
[OUTPUTS]
  HOLD OFF PRI=9,9
    AUX6 | RELAY | AHUR | AIR HANDLING UNIT RELAY 2
[END OUTPUTS]

```

Equation 8: 3rd Reset

```

[INPUTS]
  IF ANALOG = CONSTANT 3.000000 CNTS
    A259 | Compare Analog | ANALOG | ANALOG | COUNTER WHOSE INCREMENTS TURN RELAY OFF
[END INPUTS]
[OUTPUTS]
  HOLD OFF PRI=9,9
    AUX7 | RELAY | AHUR | AIR HANDLING UNIT RELAY 3
[END OUTPUTS]

```

Equation 9: 4th Reset

```

[INPUTS]
  IF ANALOG = CONSTANT 4.000000 CNTS
    A259 | Compare Analog | ANALOG | ANALOG | COUNTER WHOSE INCREMENTS TURN RELAY OFF
[END INPUTS]
[OUTPUTS]
  HOLD OFF PRI=9,9
    AUX8 | RELAY | AHUR | AIR HANDLING UNIT RELAY 4
[END OUTPUTS]

```

Equation 10: 5th Reset

[INPUTS]

IF ANALOG = CONSTANT 5.000000 CNTS

A259 | Compare Analog | ANALOG | ANALOG | COUNTER WHOSE INCREMENTS TURN RELAY OFF

[END INPUTS]

[OUTPUTS]

HOLD OFF PRI=9,9

AUX9 | RELAY | AHUR | AIR HANDLING UNIT RELAY 5

[END OUTPUTS]

Equation 11: 6th and Last Reset

[INPUTS]

IF ANALOG = CONSTANT 6.000000 CNTS

A259 | Compare Analog | ANALOG | ANALOG | COUNTER WHOSE INCREMENTS TURN RELAY OFF

[END INPUTS]

[OUTPUTS]

HOLD OFF PRI=9,9

P512 | DIGITAL | UTILITY | AHU START- UP IN PROGRESS

HOLD OFF PRI=9,9

P513 | DIGITAL | UTILITY | AHU START-UP ENABLE

HOLD OFF PRI=9,9

AUX10 | RELAY | AHUR | AIR HANDLING UNIT RELAY 6

RESET TIMER

A259 | ANALOG | ANALOG | COUNTER WHOSE INCREMENTS TURN RELAY OFF

PRINT ALL "AHU START-UP COMPLETE"

[END OUTPUTS]

Equation 12: AHU Reset Abort

[INPUTS]

STATUS ON

P3 | DIGITAL | UTILITY | FIRE ALARM DETECT

AND STATUS ON

P512 | DIGITAL | UTILITY | AHU START- UP IN PROGRESS

[END INPUTS]

[OUTPUTS]

HOLD OFF PRI=9,9

P512 | DIGITAL | UTILITY | AHU START- UP IN PROGRESS

PRINT ALL "AHU START-UP ABORTED"

[END OUTPUTS]

COUNT Example Program

This equation sets analog pseudo A440 to a value of 10 when the panel is warm started or powered up. The idea being that the value of A440 has to exceed the number of counts you wish to use. It could have a value of 100 and this would not affect the rest of the equations. In this example we are only counting up to 4 before the counter will be reset

Example:

Equation 1: SYSTEM STARTUP EQUATION

[INPUTS]

STATUS ON

A34 | ANALOG | TIMER | SYSTEM STARTUP PULSE TIMER

[END INPUTS]

[OUTPUTS]

SET CONSTANT 10 CNTS

A440 | Set-Analog (Destination) | ANALOG | ANALOG | THIS IS THE SET POINT VALUE IS 10

[END OUTPUTS]

Equation 2: SET UP COUNT EQUATION

This equation sets up the rest of the equations to use the count instruction. A441 will increment when the input is being turned on. It will maintain that count until P530 turns off and on again or if P535 turns on. At anytime during the count sequence if P535 turns on A441 will return to a value of 0.(reset)

[INPUTS]

STATUS ON

P530 | DIGITAL | UTILITY | THIS PSEUDO WILL INCREMENT COUNTER A600

COUNT ANALOG RESET ON

A441 | Analog Counter | ANALOG | ANALOG | WILL INCREMENT ON TOGGLE OF INPUT

P535 | Reset-Point | DIGITAL | UTILITY | THIS IS THE RESET PSEUDO

A440 | Analog Counter SetPoint | ANALOG | ANALOG | THIS IS THE SET POINT VALUE IS 10

[END INPUTS]

[OUTPUTS]

[END OUTPUTS]

Equation 3: FIRST COUNT EQUATION

[INPUTS]

IF ANALOG = CONSTANT 1.000000 CNTS

A441 | Compare Analog | ANALOG | ANALOG | WILL INCREMENT ON TOGGLE OF INPUT

[END INPUTS]

[OUTPUTS]

TRACK ON PRI=9,9

P531 | DIGITAL | UTILITY | TURN ON WHEN COUNTER = 1

[END OUTPUTS]

Equation 4: SECOND COUNT EQUATION

[INPUTS]

IF ANALOG = CONSTANT 2.000000 CNTS

A441 | Compare Analog | ANALOG | ANALOG | WILL INCREMENT ON TOGGLE OF INPUT

[END INPUTS]

[OUTPUTS]

TRACK ON PRI=9,9

P532 | DIGITAL | UTILITY | TURN ON WHEN COUNTER = 2

[END OUTPUTS]

Equation 5: RESET TIMER ON SYSTEM RESET

This equation will reset the counter to 0 on system reset key press (A21). It is providing the same function as turning P535 on. P535 would be controlled using custom control or a user switch.

[INPUTS]

STATUS ON

A21 | ANALOG | TIMER | SYSTEM RESET PULSE TIMER

[END INPUTS]

[OUTPUTS]

RESET TIMER

A441 | ANALOG | ANALOG | WILL INCREMENT ON TOGGLE OF INPUT

[END OUTPUTS]

Index

A

Adding Inputs - Overview	7
Adding Outputs	9
Additional Coding	85
AHU Staggered Start	88

C

COUNT Example Program	91
-----------------------------	----

E

Error Checking	11
Invalid Data	11
Syntax Error	11
Example Programming	88

I

Input Logic Operators	12
AND 1 or more points	16
AND NOT 1 or more points	16
ANY n points in a point list	28
Compare analog	29
Compare DOW value	30
COMPLEMENT current input state	19
CONTINUE current input state	18
COUNT inputs	26
CYCLE OUTPUT ACTIONS	24
DELAY output actions	22
OR 1 or more points	14
OR NOT Status 1 or more points	15
OR Status Detect 1 or more points	13
RECALL memory register	20
SAVE state to memory register	20
Introduction To Programming	4

O

Output Logic Operators	31
CALIFORNIA Code control points	62

CHLCODE Mapnet channel	48
CODE (generic) control points	64
DEVICE CODE smart sensors	49
DISABLE 1 or more points	42
DISARM 1 or more points	53
DISCONNECT 1 or more MONITOR points	41
HOLD ON 1 or more points	33
HPULSE	82
LED functions (FAST/SLOW/ON/OFF)	44
MARCH Code control points	58
MARCH120	79
MARCH20	81
MARCH60	78
Output Priority	32
PNIS Code control points	60
PRINT a message	47
PULSE analog	39
QCODE	75
RESET monitor or analog points	45
SET 1 or more points	37
SET ALARM THRESHOLD smart sensors	50
SET analog value	46
SET RATE OF RISE heat sensors	52
Signal Coding Overview	54
TCODE on a list of TrueAlert points	83, 84
TEMPORAL	80
TRACK 1 or more points	35
UPLAY or UPLAYQ audio messages	65
UPNISM message code speaker circuits	73
UPNIST tone code speaker circuits	77
UPNISV voice code speaker circuits	71
URROUTE audio	68
USILENCE audio channel	69
USPEAKER circuits on/off	70
UVOLUME	76

U

Using the Custom Control Editor	6
---------------------------------------	---